Computer Algebra for Physics Examples

Electrostatics, Magnetism, Circuits and Mechanics of Charged Particles

Part 3

Circuits

Leon Magiera and Josef Böhm

IV. Circuits

Important Laws

The current in a resistor of resistance R to which a voltage U is applied is given by *Ohm's Law*

$$I = \frac{U}{R}.$$

In more complex circuits the flow of current follows *Kirchhoff's Laws*:

Kirchoff's Current Law (KCL)

The algebraic sum of the currents flowing into and out of any junction in a circuit is equal to zero

$$\sum_k I_k = 0.$$

Kirchoff's Voltage Law (KVL)

The algebraic sum of voltages of the sources is equal to the sum of potential drops on resistors along any closed loop within a circuit.

$$\sum_k E_k = \sum_k I_k R_k.$$

An equivalent formulation is:

The algebraic sum of all voltages in a loop must equal zero.

Comments:

- The positive direction of a current flow in circuit diagrams can be arbitrarily chosen.
- Negative values for a current obtained as the solution of an exercise should be interpreted as the current flowing in the opposite direction to the assumed one.

We start with exercises in which we calculate the total circuit resistance. In the solution methods we use the following circuits properties:

- total resistance of a serial circuits equal to the sum of all resistances

$$R_T = \sum_k R_k.$$

- total conductance (reciprocal of the resistance) of a parallel circuit is the sum of the conductances of all individual branches.

$$\frac{1}{R_T} = \sum_k \frac{1}{R_k}.$$

From the above relations we see that calculation of the total resistance of serial or parallel resistors is trivial. Therefore we only calculate total resistance in serial-parallel circuits.

IV.0 Let's start with a easy problem from a textbook:

Given are: U_{01} = 30 V, U_{02} = 15 V, $R_1 = R_2$ = 200 Ω, R_3 = 400 Ω, R_4 = 600 Ω, R_5 = 300 Ω

Find the currents $I_1 - I_5$



Solution:

Following *Kirchhoff*'s laws we get a system of 5 linear equations for 5 unknowns I_1 through I_5.

(1) $U_{01} = I_1 R_1 + I_3 R_3$ (2) $U_{02} = I_2 R_2 + I_4 R_4$

(3) $I_3 R_3 = I_5 R_5 + I_4 R_4$

(4) $I_1 = I_3 + I_5$ (5) $I_4 = I_5 + I_2$

We enter the equations first, followed by fixing the parameter values. Solve gives the currents (in Ampere)

```
(%i1)  eq1:U["01"]=I[1]*R[1]+I[3]*R[3];
(eq1)  U_01=I_3 R_3+I_1 R_1
(%i2)  eq2:U["02"]=I[2]*R[2]+I[4]*R[4]$
(%i3)  eq3:I[3]*R[3]=I[5]*R[5]+I[4]*R[4]$
(%i4)  eq4:I[1]=I[3]+I[5]$
(%i5)  eq5:I[4]=I[5]+I[2]$
(%i12) U["01"]:30*V$ U["02"]:15*V$
       R[1]:200*Ω$ R[2]:200*Ω$ R[3]:400*Ω$ R[4]:600*Ω$ R[5]:300*Ω$
(%i13) V:Ω*A$
(%i14) solve([eq1,eq2,eq3,eq4,eq5],[I[1],I[2],I[3],I[4],I[5]]);
```

(%o14) $[[I_1=\frac{3\ V}{50\ \Omega}, I_2=\frac{3\ V}{400\ \Omega}, I_3=\frac{9\ V}{200\ \Omega}, I_4=\frac{9\ V}{400\ \Omega}, I_5=\frac{3\ V}{200\ \Omega}]]$

```
(%i15) %,numer;
(%o15) [[I_1=0.06 A, I_2=0.0075 A, I_3=0.045 A, I_4=0.0225 A, I_5=0.015 A]]
```

With TI-NspireCAS we can enter Volt and Ohm and we receive the currents in Ampere.

$eq1{:=}u01=i1\cdot r1+i3\cdot r3{:}eq2{:=}u02=i2\cdot r2+i4\cdot r4$	$u02=i2\cdot r2+i4\cdot r4$
$eq3{:=}i3\cdot r3=i5\cdot r5+i4\cdot r4{:}eq4{:=}i1=i3+i5{:}eq5{:=}i4=i5+i2$	$i4=i2+i5$
solve$(eq1$ and $eq2$ and $eq3$ and $eq4$ and $eq5,\{i1,i2,i3,i4,i5\})$	
$i1=\frac{r2\cdot(r3+r4+r5)\cdot u01+(r3\cdot(u01-u02)+r5\cdot u01)\cdot r4}{r1\cdot(r2\cdot(r3+r4+r5)+(r3+r5)\cdot r4)+(r2\cdot(r4+r5)+r4\cdot r5)\cdot r3}$ and $i2=\frac{r1\cdot(r3+r4+r5)\cdot u02-r3\cdot(r4\cdot(u01-u02}{r1\cdot(r2\cdot(r3+r4+r5)+(r3+r5)\cdot r4)+(r2\cdot(r4+}$	
solve$(eq1$ and $eq2$ and $eq3$ and $eq4$ and $eq5,\{i1,i2,i3,i4,i5\})\vert u01=30\cdot\_V$ and $u02=15\cdot\_V$ and $r1=200\cdot\_\Omega$ and $r2$	
$i1=0.06\cdot\_A$ and $i2=0.0075\cdot\_A$ and $i3=0.045\cdot\_A$ and $i4=0.0225\cdot\_A$ and $i5=0.015\cdot\_A$	

IV.1 Find the total resistance for the circuit presented in Fig. IV.1.



Fig IV.1

Give the numerical result for the total resistance for R_1 = 2Ω, R_2 = 3Ω, R_3 = 2Ω, R_4 = 3Ω, R_5 = 2Ω, R_6 = 3Ω, R_7 = 2Ω, R_8 = 2Ω and R_9 = 2Ω.

Solution:

Note that Resistors R_5, R_6 and R_7 are connected in series. Therefore the above circuit can be replaced by a simpler form (Fig. IV.1a).



Fig IV.1a

where $R_A = R_5 + R_6 + R_7$.

Resistors R_A and R_4 (Fig. IV.1.a) are parallel connected. Hence the next simplification is shown in Fig. IV.1.b



Fig IV.1b

where $\frac{1}{R_B} = \frac{1}{R_4} + \frac{1}{R_A}$.

As the resistors R_3, R_B and R_8 (Fig IV.1.b) are series connected we then have (Fig. IV.1c):



Fig IV.1c

with $R_C = R_3 + R_B + R_8$.

Next steps are obvious.

Resistors R_2 and R_C (Fig. IV.1.c) are parallel connected.



Fig IV.1d

where $\dfrac{1}{R_D} = \dfrac{1}{R_2} + \dfrac{1}{R_C}$.

Finally total resistance R_T can calculated according the formula

$$R_T = R_1 + R_D + R_9.$$

Let us solve the above equation system.

```
(%i5)  eq1:R[A]=R[5]+R[6]+R[7];
       eq2:1/R[B]=1/R[4]+1/R[A];
       eq3:R[C]=R[3]+R[B]+R[8];
       eq4:1/R[D]=1/R[2]+1/R[C];
       eq5:R[T]=R[1]+R[D]+R[9];
```

(eq1) $R_A = R_7 + R_6 + R_5$

(eq2) $\dfrac{1}{R_B} = \dfrac{1}{R_A} + \dfrac{1}{R_4}$

(eq3) $R_C = R_B + R_8 + R_3$

(eq4) $\dfrac{1}{R_D} = \dfrac{1}{R_C} + \dfrac{1}{R_2}$

(eq5) $R_T = R_D + R_9 + R_1$

```
(%i7)  eqs:[eq1,eq2,eq3,eq4,eq5]$
       sol:solve(eliminate(eqs,[R[A],R[B],R[C],R[D]]),R[T])[1];
```

(sol) $R_T = \big(\big((R_7+R_6+R_5+R_4) R_8 + (R_4+R_3+R_2) R_7 + (R_4+R_3+R_2) R_6 + (R_4+R_3+R_2) R_5 + (R_3+R_2) R_4\big) R_9 + \big((R_2+R_1) R_7 + (R_2+R_1) R_6 + (R_2+R_1) R_5 + (R_2+R_1) R_4\big) R_8 + \big((R_2+R_1) R_4 + (R_2+R_1) R_3 + R_1 R_2\big) R_7 + \big((R_2+R_1) R_4 + (R_2+R_1) R_3 + R_1 R_2\big) R_6 + \big((R_2+R_1) R_4 + (R_2+R_1) R_3 + R_1 R_2\big) R_5 + \big((R_2+R_1) R_3 + R_1 R_2\big) R_4 \big) / \big((R_7+R_6+R_5+R_4) R_8 + (R_4+R_3+R_2) R_7 + (R_4+R_3+R_2) R_6 + (R_4+R_3+R_2) R_5 + (R_3+R_2) R_4 \big)$

The solution of the system is quite bulky!! Finally it is easy to carry out the calculation for the given numerical data and find the resulting total resistance:

```
(%i9)  data:[R[1]=2,R[2]=3,R[3]=2,R[4]=3,R[5]=2,
       R[6]=3,R[7]=2,R[8]=2,R[9]=2]$
       subst(data,sol);
```

(%o9) $R_T = \dfrac{547}{91}$

```
(%i10) float(%);
```

(%o10) $R_T = 6.010989010989011$

With DERIVE we can do in the same way, however we miss the "eliminate" command. So we receive the bulky solution containing all variables.

#6: $$\left[RA = R5 + R6 + R7 \wedge RB = \frac{R4\cdot(R5 + R6 + R7)}{R4 + R5 + R6 + R7} \wedge RC = \frac{R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot}{R4 + R5 + R6 + R7}\right.$$

$$\frac{R2\cdot(R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5 + R6 + R7))}{R2\cdot(R4 + R5 + R6 + R7) + R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5 + R6 + R7)} \wedge RT =$$

$$\frac{R1\cdot(R2\cdot(R4 + R5 + R6 + R7) + R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5 + R6 + R7)) + R2\cdot(}{R2\cdot(R4 + R5 + R6 + R}$$

$$\frac{+ R8 + R9) + (R5 + R6 + R7)\cdot(R8 + R9)) + R9\cdot(R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5 +}{+ R7 + R8) + R8\cdot(R5 + R6 + R7)}$$

#7: $$\left[RA = 7 \wedge RB = \frac{21}{10} \wedge RC = \frac{61}{10} \wedge RD = \frac{183}{91} \wedge RT = \frac{547}{91}\right]$$

#8: $[RA = 7 \wedge RB = 2.1 \wedge RC = 6.1 \wedge RD = 2.01098901 \wedge RT = 6.01098901]$

We could also work stepwise without displaying the intermediate results:

#1: [CaseMode := Sensitive, InputMode := Word]

#2: $RA := R5 + R6 + R7$

#3: $$RB := \left(\mathrm{SOLUTIONS}\left(\frac{1}{RB} = \frac{1}{R4} + \frac{1}{RA}, RB\right)\right)_1$$

#4: $RC := R3 + RB + R8$

#5: $$RD := \left(\mathrm{SOLUTIONS}\left(\frac{1}{RD} = \frac{1}{R2} + \frac{1}{RC}, RD\right)\right)_1$$

#6: $RT = R1 + RD + R9$

#7: $RT = R1 -$

$$\frac{R3^2\cdot(R4 + R5 + R6 + R7)^2 + 2\cdot R3\cdot(R4 + R5 + R6 + R7)\cdot(R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5\sim}{(R4 + R5 + R6 + R7)\cdot(R2\cdot(R4 + R5 + R6 + R7) + \sim}$$

$$\frac{+ R6 + R7)) + R4^2\cdot(R5 + R6 + R7 + R8)^2 + 2\cdot R4\cdot R8\cdot(R5 + R6 + R7)\cdot(R5 + R6 + R7 + R8) \sim}{R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5 + R6 + R7)) \sim}$$

$$\frac{+ R8^2\cdot(R5 + R6 + R7)^2}{} +$$

$$\frac{R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8 + R9) + (R5 + R6 + R7)\cdot(R8 + R9)}{R4 + R5 + R6 + R7}$$

#8: $$RT = \frac{547}{91}$$

#9: $RT = 6.01098901$

Finally we substitute the given resistances.

If we enter the designations together with an equals-sign then we can see the intermediate results, too.

$$\#2:\quad RA := R5 + R6 + R7$$

$$\#3:\quad \left(RB := \left(\mathrm{SOLUTIONS}\left(\frac{1}{RB} = \frac{1}{R4} + \frac{1}{RA},\ RB\right)\right)\right)_1 = RB := \frac{R4\cdot(R5 + R6 + R7)}{R4 + R5 + R6 + R7}$$

$$\#4:\quad (RC := R3 + RB + R8) = RC := R3 - \frac{R5^2 + 2\cdot R5\cdot(R6 + R7) + R6^2 + 2\cdot R6\cdot R7 + R7^2}{R4 + R5 + R6 + R7} + R5 + R6 + R7 + R8$$

Below you can see the same procedure carried out with TI-NspireCAS.

$$\mathbf{ra}:=r5+r6+r7 \;\blacktriangleright\; r5+r6+r7$$

$$\mathbf{rb}:=\mathrm{right}\left(\mathrm{solve}\left(\frac{1}{\mathbf{rb}}=\frac{1}{r4}+\frac{1}{\mathbf{ra}},\mathbf{rb}\right)\right) \;\blacktriangleright\; \frac{r4\cdot(r5+r6+r7)}{r4+r5+r6+r7}$$

$$\mathbf{rc}:=r3+\mathbf{rb}+r8 \;\blacktriangleright\; r3-\frac{(r5+r6+r7)^2}{r4+r5+r6+r7}+r5+r6+r7+r8 \;⚠$$

$$\mathbf{rd}:=\mathrm{right}\left(\mathrm{solve}\left(\frac{1}{\mathbf{rd}}=\frac{1}{r2}+\frac{1}{\mathbf{rc}},\mathbf{rd}\right)\right)$$

$$\blacktriangleright\; \frac{r2\cdot\left(r3\cdot(r4+r5+r6+r7)+r4\cdot(r5+r6+r7+r8)+(r5+r6+r7)\cdot r8\right)}{r2\cdot(r4+r5+r6+r7)+r3\cdot(r4+r5+r6+r7)+r4\cdot(r5+r6+r7+r8)+(r5+r6+r7)\cdot r8} \;⚠$$

$$\mathbf{rt}:=r1+\mathbf{rd}+r9$$

$$\blacktriangleright\; r1-\frac{\left(r3\cdot(r4+r5+r6+r7)+r4\cdot(r5+r6+r7+r8)+(r5+r6+r7)\cdot r8\right)^2}{\left(r2\cdot(r4+r5+r6+r7)+r3\cdot(r4+r5+r6+r7)+r4\cdot(r5+r6+r7+r8)+(r5+r6+r7)\cdot r8\right)\cdot(r4+r5+r6+r7)}+\frac{r3\cdot(r4+r5+r6+r7)+r4\cdot(r5+r6+r7+r8+r9)+(r5+r6+r7)\cdot(r8+r9)}{r4+r5+r6+r7} \;⚠$$

$$\mathbf{rt}|r1=2 \text{ and } r2=3 \text{ and } r3=2 \text{ and } r4=3 \text{ and } r5=2 \text{ and } r6=3 \text{ and } r7=2 \text{ and } r8=2 \text{ and } r9=2$$

$$\blacktriangleright\; \mathbf{6.01099}$$

The reader should have no trouble solving similar problems.

■

IV.2 Calculate the resistance R_{AB} equivalent to the system of resistors R_1, …, R_5 connected as shown in Fig. IV.2.



Fig. IV.2

Solution:

Applying *Kirchhoff*'s law we obtain the following system of equations:

$I\, R_{AB} = I_1\, R_1 + I_2\, R_2$

$I_1\, R_1 + (I_1 - I_2) - (I - I_1)\, R_3 = 0$ for the sub-circuit ACDA

$I_2\, R_2 + (I - I_2) - (I_1 - I_2)\, R_5 = 0$ for the sub-circuit CBDC

It remains to enter the equations and solve the system for R_{AB}:

```
(%i1)  eq1:I*R[AB]=I[1]*R[1]+I[2]*R[2];
```
(eq1) $R_{AB}\, I = I_2\, R_2 + I_1\, R_1$

```
(%i2)  eq2:I[1]*R[1]+(I[1]-I[2])*R[5]-(I-I[1])*R[3]=0;
```
(eq2) $R_3\,(I_1 - I) + (I_1 - I_2)\, R_5 + I_1\, R_1 = 0$

```
(%i3)  eq3:I[2]*R[2]-(I-I[2])*R[4]-(I[1]-I[2])*R[5]=0;
```
(eq3) $-R_4\,(I - I_2) + (I_2 - I_1)\, R_5 + I_2\, R_2 = 0$

```
(%i4)  RAB:solve(eliminate([eq1,eq2,eq3],[I[1],I[2]]),R[AB]);
```
(RAB) $$\left[R_{AB} = \frac{((R_2+R_1)\,R_4 + (R_2+R_1)\,R_3)\,R_5 + ((R_2+R_1)\,R_3 + R_1\,R_2)\,R_4 + R_1\,R_2\,R_3}{(R_4+R_3+R_2+R_1)\,R_5 + (R_3+R_1)\,R_4 + R_2\,R_3 + R_1\,R_2} \right]$$

With TI-NspireCAS we cannot eliminate the auxiliary currents, so we have to display the whole solution. Substituting the same data set makes sure that both results are identical.

$eq1 := i \cdot rab = i1 \cdot r1 + i2 \cdot r2 : eq2 := i1 \cdot r1 + (i1 - i2) \cdot r5 - (i - i1) \cdot r3 = 0$

$-i \cdot r3 + i1 \cdot (r1 + r3 + r5) - i2 \cdot r5 = 0$

$eq3 := i2 \cdot r2 - (i - i2) \cdot r4 - (i1 - i2) \cdot r5 = 0$ $\quad -i \cdot r4 - i1 \cdot r5 + i2 \cdot (r2 + r4 + r5) = 0$

solve$(eq1$ and $eq2$ and $eq3, \{rab, i1, i2\})$

$\frac{r5)}{+r5)+r4 \cdot r5}$ and $rab = \frac{r1 \cdot (r2 \cdot (r3+r4) + r3 \cdot (r4+r5) + r4 \cdot r5) + r2 \cdot (r3 \cdot (r4+r5) + r4 \cdot r5)}{r1 \cdot (r2+r4+r5) + r2 \cdot (r3+r5) + r3 \cdot (r4+r5) + r4 \cdot r5}$

$\frac{r1 \cdot (r2 \cdot (r3+r4) + r3 \cdot (r4+r5) + r4 \cdot r5) + r2 \cdot (r3 \cdot (r4+r5) + r4 \cdot r5)}{r1 \cdot (r2+r4+r5) + r2 \cdot (r3+r5) + r3 \cdot (r4+r5) + r4 \cdot r5} | r1 = 1$ and $r2 = 2$ and $r3 = 3$

$\frac{155}{74}$

```
(%i5)   subst([R[1]=1,R[2]=2,R[3]=3,R[4]=4,R[5]=5],RAB);
```
$$(\%o5)\quad [R_{AB}=\frac{155}{74}]$$

For calculating the total resistance for more complex circuits it is profitable to apply Y-Δ and Δ-Y Conversions. These conversions are explained in the next problem.

IV.3 Three resistors of resistances R_1, R_2, and R_3 are connected in form of a "star" (**Y**, right in Fig. IV.3), whilst the resistors R_{12}, R_{23}, R_{13} in form of a triangle (**Δ**, left in Fig. IV.3). Calculate the resistances in the (**Y**) as a function of the resistances in the (**Δ**) and vice versa, if it known that the resistances between two equivalent nodes in the two circuits are identical.



Fig. IV.3

Solution:

The following equations follow from the equivalence condition:

```
(%i1)   eq1:R[1]+R[3]=1/(1/R[13]+1/(R[12]+R[23]));
```
$$(eq1)\quad R_3+R_1=\frac{1}{\frac{1}{R_{23}+R_{12}}+\frac{1}{R_{13}}}$$

```
(%i2)   eq2:R[1]+R[2]=1/(1/R[12]+1/(R[13]+R[23]));
```
$$(eq2)\quad R_2+R_1=\frac{1}{\frac{1}{R_{23}+R_{13}}+\frac{1}{R_{12}}}$$

```
(%i3)   eq3:R[2]+R[3]=1/(1/R[23]+1/(R[13]+R[12]));
```
$$(eq3)\quad R_3+R_2=\frac{1}{\frac{1}{R_{23}}+\frac{1}{R_{13}+R_{12}}}$$

We solve it:

- first with respect to the unknowns R_1, R_2, R_3 (**Δ-Y** conversion)

```
(%i4)   solve([eq1,eq2,eq3],[R[1],R[2],R[3]]);
```
$$(\%o4)\quad [[R_1=\frac{R_{12}\,R_{13}}{R_{23}+R_{13}+R_{12}},\ R_2=\frac{R_{12}\,R_{23}}{R_{23}+R_{13}+R_{12}},\ R_3=\frac{R_{13}\,R_{23}}{R_{23}+R_{13}+R_{12}}]]$$

Each resistor of the **Y**-connection is equal to the product of the resistors in the two closest branches of the **Δ**-connection divided by the sum of the resistors in Δ.

- then with respect to the unknowns R_{12}, R_{23}, R_{13} (Y-Δ conversion)

```
(%i5)   solve([eq1,eq2,eq3],[R[12],R[23],R[13]]);
```

(%o5) $[[R_{12}=0, R_{23}=0, R_{13}=0], [R_{12}=\frac{(R_2+R_1)R_3+R_1R_2}{R_3}, R_{23}=\frac{(R_2+R_1)R_3+R_1R_2}{R_1}, R_{13}=\frac{(R_2+R_1)R_3+R_1R_2}{R_2}]]$

Each resistor of the **Δ**-connection is equal to the sum of the possible product combinations of the **Y**-connection divided by the resistance of the **Y** farthest from the resistor to be determined.

Let's now solve two problems by applying the above equivalence condition.

IV.4 Evaluate the total resistance between nodes A and B given in Fig. IV.4a.
Show that the total resistance between nodes A and B equals *r*, if all appearing resistances are identical and equal *r*.



Fig. IV.4a

Solution:

Note that the Δ-configuration of resistors R_1, R_2 and R_5 (Fig. IV.4a) can be converted to a **Y**-configuration (Fig.IV4b).



Fig. IV.4b

We use the solution of problem IV.3. Hence the assignments as follows:

```
(%i3)   R[A]:R[1]*R[2]/(R[1]+R[2]+R[5]);
        R[B]:R[1]*R[5]/(R[1]+R[2]+R[5]);
        R[D]:R[2]*R[5]/(R[1]+R[2]+R[5]);
```

(%o1) $\dfrac{R_1\,R_2}{R_5+R_2+R_1}$

(%o2) $\dfrac{R_1\,R_5}{R_5+R_2+R_1}$

(%o3) $\dfrac{R_2\,R_5}{R_5+R_2+R_1}$

As the pairs of resistors R_B, R_3 and R_D, R_4 are in series, the network of Fig. IV4b can be transformed to the simpler form given below (Fig. IV4c).



Fig. IV4c

These are the assignments for the resistors R_{B3} and R_{D4}.

```
(%i5)   R[B3]:R[B]+R[3];
        R[D4]:R[D]+R[4];
```

(%o4) $\dfrac{R_1\,R_5}{R_5+R_2+R_1}+R_3$

(%o5) $\dfrac{R_2\,R_5}{R_5+R_2+R_1}+R_4$

Resistors R_{B3} and R_{D4} are parallel connected. This connection and R_A are in series which leads to the following two equations.

```
(%i6)   eq1:1/R[B3D4]=1/R[B3]+1/R[D4];
```

(eq1) $\dfrac{1}{R_{B3D4}}=\dfrac{1}{\dfrac{R_2\,R_5}{R_5+R_2+R_1}+R_4}+\dfrac{1}{\dfrac{R_1\,R_5}{R_5+R_2+R_1}+R_3}$

```
(%i7)   eq2:R[AC]=R[A]+R[B3D4];
```

(eq2) $R_{AC}=R_{B3D4}+\dfrac{R_1\,R_2}{R_5+R_2+R_1}$

Now we can find the generic solution for R_{AC}:

```
(%i8)   sol:solve(eliminate([eq1,eq2],[R[B3D4]]),R[AC]);
```

(sol) $[\,R_{AC}=\dfrac{((R_3+R_1)\,R_4+R_2\,R_3+R_1\,R_2)\,R_5+((R_2+R_1)\,R_3+R_1\,R_2)\,R_4+R_1\,R_2\,R_3}{(R_4+R_3+R_2+R_1)\,R_5+(R_2+R_1)\,R_4+(R_2+R_1)\,R_3}\,]$

Finally simplifying the resistance R_{AC} for identical values for all given resistances we get:

```
(%i9)   subst([R[1]=r,R[2]=r,R[3]=r,R[4]=r,R[5]=r],sol);
(%o9)   [R_AC=r]
```

$$[R_{AC}=r]$$

It is interesting that even for $R_1 = R_2 = R_3 = R_4 = r$ it turns out that R_{AC} is also r.

```
(%i10)  subst([R[1]=r,R[2]=r,R[3]=r,R[4]=r],sol);
```

(%o10) $$\left[R_{AC}=\frac{4\,r^3+4\,R_5\,r^2}{4\,r^2+4\,R_5\,r}\right]$$

```
(%i11)  ratsimp(%);
(%o11)  [R_AC=r]
```

This says that resistance R_{AC} does not depend on the central resistor with resistance R_5.

Exercise: Repeat the calculation by applying the **Δ-Y** conversion on the configuration of resistors R_3, R_4 and R_5.

The TI-Nspire solution is given below.

$$\mathbf{re}:=\frac{r3\cdot r4}{r3+r4+r5} \quad \mathbf{rg}:=\frac{r3\cdot r5}{r3+r4+r5} \quad \mathbf{rf}:=\frac{r4\cdot r5}{r3+r4+r5}$$

$$\mathbf{rg1}:=\mathbf{rg}+r1 \quad \mathbf{rf2}:=\mathbf{rf}+r2$$

$$\mathbf{eq3}:=\frac{1}{rg1f2}=\frac{1}{\mathbf{rg1}}+\frac{1}{\mathbf{rf2}}$$

Don't use two equations - solving the system needs too much time!

$$\mathbf{sol}:=\text{right}(\text{solve}(\mathbf{eq3},rg1f2)) \blacktriangleright \frac{(r1\cdot(r3+r4+r5)+r3\cdot r5)\cdot(r2\cdot(r3+r4+r5)+r4\cdot r5)}{(r1\cdot(r3+r4+r5)+r2\cdot(r3+r4+r5)+(r3+r4)\cdot r5)\cdot(r3+r4+r5)}$$

$$\mathbf{rac}:=\text{comDenom}(\mathbf{re}+\mathbf{sol})$$

$$\blacktriangleright \frac{r1\cdot r2\cdot r3+r1\cdot r2\cdot r4+r1\cdot r2\cdot r5+r1\cdot r3\cdot r4+r1\cdot r4\cdot r5+r2\cdot r3\cdot r4+r2\cdot r3\cdot r5+r3\cdot r4\cdot r5}{r1\cdot r3+r1\cdot r4+r1\cdot r5+r2\cdot r3+r2\cdot r4+r2\cdot r5+r3\cdot r5+r4\cdot r5}$$

Substitute r for all resistances:

$$\mathbf{rac}|r1=r \text{ and } r2=r \text{ and } r3=r \text{ and } r4=r \text{ and } r5=r \blacktriangleright r$$

$$\mathbf{rac}|r1=r \text{ and } r2=r \text{ and } r3=r \text{ and } r4=r \blacktriangleright r$$

IV.5 Calculate the resistance "observed" between points 1 and 2 of the circuit presented in Fig. IV.5. The resistances denoted on the diagram are assumed to be known.



Fig. IV.5

Solution:

In order to calculate the required resistance, we make use of the transformations of a Δ into a **Y** and vice versa derived in the previous example. Each resistor of **Y** (see Problem IV.3) is equal to the product of the resistors in the two closest branches of the Δ divided by the sum of the resistors in the **Δ.** We replace the 3-4-5 Δ by a **Y** with resistances *a_1, a_2, a_3*. The equivalent circuit resulting from such a transformation is presented in Fig. IV.5a (right figure with intermediate step left).



Fig. IV.5a

$$a_1 = a_2 = \frac{r \cdot r_{35}}{r + r + r_{35}} = \frac{r \cdot r_{35}}{2r + r_{35}}, \quad a_3 = \frac{r \cdot r}{r + r + r_{35}} = \frac{r^2}{2r + r_{35}}$$

After a subsequent transformation from the 1-2-4 **Y** of the resistances r, r and a_3 into a Δ formed by b_1, b_2 and b_3 which can be presented a shown in Fig. IV.5b (right).



Fig. IV.5b

Each resistor of the Δ (see Problem IV.3) is equal to the sum of the possible product combination of the resistances of the **Y** divided by the resistance of the **Y** farthest from the resistor to be determined. So we get the resistances b_1, b_2, b_3

$$b_1 = b_2 = a_3 + r + \frac{a_3 \cdot r}{r} = 2a_3 + r, \quad b_3 = r + r + \frac{r \cdot r}{a_3} = 2r + \frac{r^2}{a_3}$$

Circuit IV.5b can finally be transformed into a simple one which is shown in Fig. IV.5c:



Fig. IV.5c

$$\frac{1}{c_1} = \frac{1}{r_{15} + a_1} + \frac{1}{b_1}, \quad \frac{1}{c_2} = \frac{1}{r_{23} + a_2} + \frac{1}{b_2}, \quad \frac{1}{c_3} = \frac{1}{r_{12}} + \frac{1}{b_3}$$

The resultant resistance c of the given circuit is given by the formula (according to Kirchhoff's laws $\frac{1}{c} = \frac{1}{c_1 + c_2} + \frac{1}{c_3}$.

All equations deduced above can easily be transferred to the CAS. Maxima will solve the system of equations for the requested resistance c.

```
(%i3)   a[1]:r*r[35]/(2*r+r[35])$ a[2]:a[1]$  a[3]:r^2/(2*r+r[35])$
(%i6)   b[1]:2*a[3]+r$  b[2]:b[1]$  b[3]:2*r+r^2/a[3]$
(%i9)   eq1:1/c[1]=1/(r[15]+a[1])+1/b[1]$
        eq2:1/c[2]=1/(r[23]+a[2])+1/b[2]$
        eq3:1/c[3]=1/r[12]+1/b[3]$
(%i10)  eq:1/c=1/(c[1]+c[2])+1/c[3]$
(%i11)  sol:solve(eliminate([eq1,eq2,eq3,eq],[c[1],c[2],c[3]]),c)$
```

We can close with a numerical example entering data for the given resistances:

```
 Numerical Example:
(%i14)  r[12]:r[15]:r[23]:r[35]:3*Ω$
        r:2*Ω$
        ev(sol);
```

(%o14) $\left[c = \frac{108\,\Omega}{77}\right]$

```
(%i15)  bfloat(sol),fpprec:4;
(%o15)  [c=1.403b0 Ω]
```

The solution of the next problem will be used in the further part of this chapter.

IV. 6 Calculate the resistance between points A and B equivalent to an infinitely long circuit consisting of resistances R_1 and R_2 (Fig. IV.6a).



Fig. IV.6a

Solution:

The circuit illustrated above can be simplified to the following form;



Fig. IV.6b

The value of the resistance of such a circuit consisting of an infinite number of elements cannot change when an extra element is added:



Fig. IV.6c

Thus, we obtain the equation: $R = R_1 + \dfrac{1}{\dfrac{1}{R_2} + \dfrac{1}{R}}$.

We solve this equation for R:

(%i1) `sol:solve(R=R[1]+1/(1/R[2]+1/R),R);`

(sol) $\left[R = -\dfrac{\sqrt{4\,R_1\,R_2 + R_1^2} - R_1}{2},\ R = \dfrac{\sqrt{4\,R_1\,R_2 + R_1^2} + R_1}{2} \right]$

(%i2) `sol[2];`

(%o2) $R = \dfrac{\sqrt{4\,R_1\,R_2 + R_1^2} + R_1}{2}$

Ignoring the negative solution we receive the final solution for R.

See also problem IV.22 and IV.23.

■━━━━━━━━━━

The next problem leads to a system with more unknowns than equations.

IV. 7 Calculate the electromotive force E and the internal resistance r of the one cell equivalent to two cells of electromotive forces E_1 and E_2 and internal resistances r_1 and r_2 (see Fig IV.7a and IV.7b), so that the current flowing through resistance R is the same, regardless of the value of this resistance.



Fig. IV.7a



Fig. IV.7b

Solution:

First we write the equations for both circuits applying *Ohm*'s and *Kirchhoff*'s laws:

- for circuit (a)

$$E_1 = I_1 r_1 + I\,R$$
$$E_2 = I_2 r_2 + I\,R$$
$$I = I_1 + I_2$$

- for circuit (b)

$$E = I\,(R + r)$$

Thus, we have five unknowns I, I_1, I_2, R and r, but only four equations. The question is howto solve the problem.

Let's enter the equations in our computer algebra system.

```
(%i4)   eq1:E[1]=I[1]*r1+I*R$     eq2:E[2]=I[2]*r2+I*R$
        eq3:I=I[1]+I[2]$          eq4:E=I*(R+r)$
```

In the fist step we will treat I, I_1 and I_2 as unknowns and solve the first three equations only for current I. In the second step we will solve the fourth equation for I, too.

```
(%i6)   eqs:[eq1,eq2,eq3]$
        solve(eliminate(eqs,[I[1],I[2]]),I);
```

(%o6) $\left[I = \dfrac{E_1\,r2 + E_2\,r1}{(r2+r1)\,R + r1\,r2}\right]$

```
(%i7)   solve(eq4,I);
```

(%o7) $\left[I = \dfrac{E}{R+r}\right]$

As the current should be the same, we equate both expressions for I, receiving one equation for the remaining two unknowns, R and r.

```
(%i8)   (E[1]*r2+E[2]*r1)/((r2+r1)*R+r1*r2)=E/(R+r);
```

(%o8) $\dfrac{E_1\,r2 + E_2\,r1}{(r2+r1)\,R + r1\,r2} = \dfrac{E}{R+r}$

We rearrange this equation by bringing them over the common denominator:

```
(%i9)   eq:%o8*((r2+r1)*R+r1*r2)*(R+r);
(eq)    (E_1 r2+E_2 r1)(R+r)=E((r2+r1) R+r1 r2)
```

The coefficients of R and the constant terms as well on both sides of the equation must be equal. This leads to another two equations, which then can be solved for E and r.

```
(%i11)  eq5:coeff(expand(eq),R);eq6:coeff(expand(eq),R,0);
(eq5)   E_1 r2+E_2 r1=r2 E+r1 E
(eq6)   E_1 r r2+E_2 r r1=r1 r2 E
(%i12)  solve([eq5,eq6],[E,r]);
(%o12)  [[E=(E_1 r2+E_2 r1)/(r2+r1), r=(r1 r2)/(r2+r1)]]
```

You can easily follow the DERIVE treatment of problem IV.7

```
#1:   [CaseMode := Sensitive, InputMode := Word]
#2:   [eq1 := E1 = I1·r1 + I·R, eq2 := E2 = I2·r2 + I·R]
#3:   [eq3 := I = I1 + I2, eq4 := E = I·(R + r)]
#4:   (I_ := (SOLUTIONS([eq1, eq2, eq3], [I1, I2, I]))_1,3 ) = I_ := (E1·r2 + E2·r1)/(R·(r1 + r2) + r1·r2)
#5:   (I__ := (SOLUTIONS(eq4, I))_1 ) = I__ := E/(R + r)
#6:   (eq := I_ = I__) = eq := (E1·r2 + E2·r1)/(R·(r1 + r2) + r1·r2) = E/(R + r)
#7:   eq_ := eq·(R·(r1 + r2) + r1·r2)·(R + r)
#8:   eq_ := (R + r)·(E1·r2 + E2·r1) = E·(R·(r1 + r2) + r1·r2)
#9:   eq5 := POLY_COEFF(LHS(eq_), R, 1) = POLY_COEFF(RHS(eq_), R, 1)
#10:  eq5 := E1·r2 + E2·r1 = E·(r1 + r2)
#11:  eq6 := POLY_COEFF(LHS(eq_), R, 0) = POLY_COEFF(RHS(eq_), R, 0)
#12:  eq6 := r·(E1·r2 + E2·r1) = E·r1·r2
#13:  SOLVE(eq5 ∧ eq6, [E, r]) = [E = (E1·r2 + E2·r1)/(r1 + r2) ∧ r = r1·r2/(r1 + r2)]
```

IV. 8 Calculate the current and the drop in potential at the external resistors in each arm of the circuit given in Fig. IV.8. The electromotive forces are equal to E_1, E_2 and their internal resistances are equal to r_1 and r_2. The external resistances R_1, R_2, R_3 are also known.



Fig. IV.8

Solution:

Applying *Kirchhoff*'s laws we write, in accordance with the notation given in Fig. IV.8 the following set of equations:

$$I_3 = I_1 + I_2$$
$$E_1 = I_1\,(R_1 + r_1) + I_3\,R_3$$
$$E_2 = I_2\,(R_2 + r_2) + I_3\,R_3$$

The drops in potential are

$$U_1 = I_1\,R_1$$
$$U_2 = I_2\,R_2$$
$$U_3 = I_3\,R_3$$

We have two possibilities to find all unknowns: enter all six equations and solve the system for I_1 to U_3 or solve the first three equations for I_1, I_2 and I_3 and then substitute for getting the potential drops.

We choose the first way:

```
(%i6)   eq1:I3=I1+I2$   eq2:E1=I1*(R1+r1)+I3*R3$
        eq3:E2=I2*(R2+r2)+I3*R3$
        eq4:U1=I1*R1$   eq5:U2=I2*R2$   eq6:U3=I3*R3$

(%i7)   solve([eq1,eq2,eq3,eq4,eq5,eq6],[I1,I2,I3,U1,U2,U3]);
```

(%o7)
$$[[I1=\frac{(E1-E2)\,R3+E1\,(R2+r2)}{(R2+R1+r2+r1)\,R3+R1\,(R2+r2)+r1\,(R2+r2)},\ I2=-\frac{(E1-E2)\,R3-E2\,R1-r1\,E2}{(R2+R1+r2+r1)\,R3+R1\,(R2+r2)+r1\,(R2+r2)},\ I3=\frac{E1\,(R2+r2)+E2\,R1+r1\,E2}{(R2+R1+r2+r1)\,R3+R1\,(R2+r2)+r1\,(R2+r2)},$$
$$U1=\frac{(E1-E2)\,R1\,R3+E1\,R1\,(R2+r2)}{(R2+R1+r2+r1)\,R3+R1\,(R2+r2)+r1\,(R2+r2)},\ U2=-\frac{(E1\,R2-E2\,R2)\,R3-E2\,R1\,R2-r1\,E2\,R2}{(R2+R1+r2+r1)\,R3+R1\,(R2+r2)+r1\,(R2+r2)},\ U3=\frac{\left(E1\,(R2+r2)+E2\,R1+r1\,E2\right)R3}{(R2+R1+r2+r1)\,R3+R1\,(R2+r2)+r1\,(R2+r2)}]]$$

IV. 9 The two cells are connected by a resistor of resistance R as shown in Fig. IV.8. Electromotive forces of these cells are E_1 and E_2 and their internal resistances are r_1 and r_2, respectively. Calculate the currents in all the branches of the circuit.



Fig. IV.9

Solution:

We provide two solution methods.

Method 1:

From *Kirchhoff*'s laws we obtain the following equations:

for node A: $I_1 + I_2 = I_3$

for sub-circuit *ECDF*: $E_1 - E_2 = I_1\, r_1 - I_2\, r_2$

for sub-circuit *EABF* $E_1 = I_1\, r_1 + I_3\, R$

We solve this linear system for the currents. Then we enter a set of numerical data for a special circuit.

```
(%i3)   eq1:I[1]+I[2]=I[3]$
        eq2:E[1]-E[2]=I[1]*r[1]-I[2]*r[2]$
        eq3:E[1]=I[1]*r[1]+I[3]*R$
(%i4)   sol:solve([eq1,eq2,eq3],[I[1],I[2],I[3]]);
```

(sol) $$[[I_1 = -\frac{E_2\, R - E_1\, R - E_1\, r_2}{r_2\,(R+r_1) + r_1\, R},\ I_2 = \frac{E_2\,(R+r_1) - E_1\, R}{r_2\,(R+r_1) + r_1\, R},\ I_3 = \frac{r_1\, E_2 + E_1\, r_2}{r_2\,(R+r_1) + r_1\, R}]]$$

Numerical data: $R = 0.6\Omega$, $E_1 = 1.6\text{V}$, $E_2 = 1.2\text{V}$, $r_1 = 1\Omega$, $r_2 = 0.5\Omega$.
We use the relationship $\text{V} = \text{A} * \Omega$.

```
(%i5)   subst([r[1]=Ω, r[2]=0.5*Ω, R=0.6*Ω,
              E[1]=1.6*V,E[2]=1.3*V,V=A*Ω],sol);
(%o5)   [[I[1]=0.7000000000000002 A, I[2]=0.8 A, I[3]=1.5 A]]
```

Exercise: In our solution we used sub-circuit *EABF:* What do you think will happen when using sub-circuit *ACDB* instead? Confirm your conjecture.

Method 2:

We apply the method of nodal potentials. Let the potentials at node A and B be φ_A and φ_B, respectively. We can assume $\varphi_B = 0$. The voltage U_{AB} between points A and B is

$$U_{AB} = \varphi_A - \varphi_B = \varphi_A$$

Using *Ohm*'s and *Kirchhoff*'s laws we can write:

$$I_1 = \frac{E_1 - U_{AB}}{r_1},\quad I_2 = \frac{E_2 - U_{AB}}{r_2},\quad I_3 = \frac{U_{AB}}{R},$$

$$I_1 + I_2 = I_3$$

We enter these equations

```
(%i1) [eq1:I1+I2=I3,
       eq2:I1=(E1-UAB)/r1,
       eq3:I2=(E2-UAB)/r2,
       eq4:I3-UAB/R]$
```

and solve them

```
(%i2) sol:solve(
            eliminate([eq1,eq2,eq3,eq4],[UAB]),
          [I1,I2,I3]);
```

$$\text{(\%o2)}\quad [[I1=\frac{(E1-E2)\ R+r2\ E1}{(r2+r1)\ R+r1\ r2}, I2=-\frac{(E1-E2)\ R-r1\ E2}{(r2+r1)\ R+r1\ r2}, I3=\frac{r1\ E2+r2\ E1}{(r2+r1)\ R+r1\ r2}]]$$

```
(%i3) bfloat(
           subst([r1=1,r2=0.5,R=0.6,E1=1.6,E2=1.3],sol)),
      fpprec:3;
(%o3) [[I1=7.0b-1,I2=8.0b-1,I3=1.5b0]]
```

As should be expected the obtained solution is identical to the result obtained using *Method I.*

IV. 10 Calculate the current flowing in all the branches of the circuit presented in Fig. IV.9. The following Data are given: R_1 = 12Ω, R_1 = 12Ω, R_2 = 7Ω, R_3 = 6Ω, R_4 = 5Ω, R_5 = 11Ω, R_6 = 8Ω, E_1 = 2V, E_2 = 3V, E_3 = 1.5V, E_4 = 2.5V, E_5 = 3V.



Fig. IV.10

Solution:

The equations following *Kirchhoff*'s laws are of the form:

First we write the equations for both circuits applying *Ohm*'s and *Kirchhoff*'s laws:

- for the nodes

$I_5 = I_1 + I_3$

$I_4 = I_3 - I_2$

$I_6 = I_1 + I_2$

- for the sub-circuits

$E_1 + E_5 = R_1 I_1 + R_5 I_5 + R_6 I_6$

$E_3 - E_4 = R_2 I_2 - R_4 I_4 + R_6 I_6$

$E_3 + E_4 + E_5 = R_3 I_3 + R_4 I_4 + R_5 I_5$

Let's solve the system for the currents considering the given data – without displaying the general solution.

```
(%i1)   data:[E[1]:2,E[2]:3,E[3]:1.5,E[4]:2.5,E[5]:3,
              R[1]:12,R[2]:7,R[3]:6, R[4]:5,R[5]:11,R[6]:8]$

(%i7)   eq1:I[5]=I[1]+I[3]$  eq2:I[4]=I[3]-I[2]$  eq3:I[6]=I[1]+I[2]$
        eq4:E[1]+E[5]=R[1]*I[1]+R[5]*I[5]+R[6]*I[6]$
        eq5:E[2]-E[4]=R[2]*I[2]-R[4]*I[4]+R[6]*I[6]$
        eq6:E[3]+E[4]+E[5]=R[3]*I[3]+R[4]*I[4]+R[5]*I[5]$

(%i10)  eqs:[eq1,eq2,eq3,eq4,eq5,eq6]$
        unkn:[I[1],I[2],I[3],I[4],I[5],I[6]]$
        solve(eqs,unkn);
rat: replaced 0.5 by 1/2 = 0.5
rat: replaced 7.0 by 7/1 = 7.0
```

(%o10) $[[I_1=\frac{93}{5438}, I_2=\frac{551}{5438}, I_3=\frac{1809}{5438}, I_4=\frac{629}{2719}, I_5=\frac{951}{2719}, I_6=\frac{322}{2719}]]$

```
(%i11)  %, numer;
(%o11)  [[I1=0.01710187568959176,I2=0.10132401618242,I3=0.3326590658330268,
I4=0.2313350496506068,I5=0.3497609415226186,I6=0.1184258918720118]]

(%i12)  bfloat(%),fpprec:3;
(%o12)  [[I1=1.71b-2,I2=1.01b-1,I3=3.33b-1,I4=2.31b-1,I5=3.5b-1,I6=1.18b-1]
]
```

Notice the different forms of the numerical results:

- as rational number in %o10
- as a long decimal number in %o11 (16 significant digits)
- as a floating point decimal number, e.g. 1.17b–2 = $1.17 \cdot 10^{-2}$

IV. 11 Calculate the current in all arms of the circuit presented in the bridge circuit (Fig. IV.11). Bridge circuit is a configuration which has a lot of applications (see https://en.wikipedia.org/wiki/Bridge_circuit).

The values of voltages and resistances denoted in the diagram are as follows: R_1 = 13Ω, R_2 = 5Ω, R_3 = 18Ω, R_4 = 15Ω, R_5 = 10Ω, R_6 = 13Ω, E_1 = 2V, E_2 = 2.5V, E_3 = 1.2V, E_5 = 2V, E_6 = 3V.



Fig. IV.11

Solution:

We will present two methods for solving this problem.

Method 1:

As in the previous problem we set up a system of equations satisfying *Kirchhoff*'s laws:

- for the nodes

$I_4 = I_1 + I_2$

$I_5 = I_2 - I_3$

$I_6 = I_1 + I_3$

- for the sub-circuits

$E_1 + E_6 = R_1 I_1 + R_4 I_4 + R_6 I_6$

$E_2 + E_5 = R_2 I_2 + R_4 I_4 + R_5 I_5$

$E_3 - E_5 + E_6 = R_3 I_3 - R_5 I_5 + R_6 I_6$

```
(%i1)   data:[E[1]:E[5]:2,E[2]:2.5,E[3]:1.2,E[6]:3,
               R[1]:R[6]:13,R[2]:5,R[3]:18, R[4]:15,R[5]:10]$
(%i7)   eq1:I[4]=I[1]+I[2]$  eq2:I[5]=I[2]-I[3]$
        eq3:I[6]=I[1]+I[3]$
        eq4:E[1]+E[6]=R[1]*I[1]+R[4]*I[4]+R[6]*I[6]$
        eq5:E[2]+E[5]=R[2]*I[2]+R[4]*I[4]+R[5]*I[5]$
        eq6:E[3]-E[5]+E[6]=R[3]*I[3]-R[5]*I[5]+R[6]*I[6]$
(%i10)  eqs:[eq1,eq2,eq3,eq4,eq5,eq6]$
        unkn:[I[1],I[2],I[3],I[4],I[5],I[6]]$
        solve(eqs,unkn);
rat: replaced 4.5 by 9/2 = 4.5
rat: replaced 2.2 by 11/5 = 2.2
```

(%o10) $[[I_1=\frac{2219}{56270}, I_2=\frac{882}{5627}, I_3=\frac{4467}{56270}, I_4=\frac{11039}{56270}, I_5=\frac{4353}{56270}, I_6=\frac{3343}{28135}]]$

```
(%i11)  bfloat(%),fpprec:3;
```

(%o11) $[[I_1=3.94b-2, I_2=1.57b-1, I_3=7.94b-2, I_4=1.96b-1, I_5=7.74b-2, I_6=1.19b-1]]$

Method 2:

We apply the method of sub-circuits (meshes). This method enables the writing of $n - m + 1$ (= number of meshes) linear equations for a circuit with n branches and m nodes in the form

$$\sum_{j=1}^{number\ of\ meshes} R_{ij} \cdot I_{0j} = E_{0i}, \qquad (i = 1,\ 2,\ \ldots,\ n - m + 1)$$

where: R_{ij} – the resultant resistance of the common branch of sub-circuits i and j ($i \neq j$, $R_{ij} < 0$ if the assumed directions of current are opposite, $R_{ij} > 0$ if they are the same); R_{ii} – the resistance of the i-th sub-circuit; I_{0i} – current in the i-th sub-circuit; E_{0i} – resultant electric potential of the i-th sub-circuit. To avoid misunderstanding between I_{01} and I_1 we write Ia, Ib, Ic for I_{01}, I_{02}, I_{03} and Ea, Eb, Ec for E_{01}, E_{02}, E_{03}. We obtain the following system of linear equations for the three meshes:

$(R_1 + R_4 + R_6)\ Ia + R_4\ Ib + R_6\ Ic = E_1 + E_6 \quad \leftrightarrow \quad R_{11}\ Ia + R_{12}\ Ib + R_{13}\ Ic = E_{01} = Ea$

$R_4\ Ia + (R_2 + R_4 + R_5)\ Ib - R_5\ Ic = E_2 + E_5 \quad \leftrightarrow \quad R_{21}\ Ia + R_{22}\ Ib + R_{23}\ Ic = E_{02} = Eb$

$R_6\ Ia - R_5\ Ib + (R_3 + R_5 + R_6)\ Ic = E_3 - E_5 + E_6 \quad \leftrightarrow \quad R_{31}\ Ia + R_{32}\ Ib + R_{33}\ Ic = E_{03} = Ec$

This system and its solution can be written in matrix form:

$$\begin{bmatrix} R_{11} & R_{12} & R_{13} \\ R_{21} & R_{22} & R_{23} \\ R_{31} & R_{32} & R_{33} \end{bmatrix} * \begin{bmatrix} Ia \\ Ib \\ Ic \end{bmatrix} = \begin{bmatrix} Ea \\ Eb \\ Ec \end{bmatrix} \rightarrow \begin{bmatrix} Ia \\ Ib \\ Ic \end{bmatrix} = \begin{bmatrix} R_{11} & R_{12} & R_{13} \\ R_{21} & R_{22} & R_{23} \\ R_{31} & R_{32} & R_{33} \end{bmatrix}^{-1} * \begin{bmatrix} Ea \\ Eb \\ Ec \end{bmatrix} \text{ or short } I = R^{-1} \cdot E.$$

Now see the Maxima-realization which fortunately gives the same results as method 1:

```
(%i1)   data:[E[1]:E[5]:2,E[2]:2.5,E[3]:1.2,E[6]:3,
              R[1]:R[6]:13,R[2]:5,R[3]:18, R[4]:15,R[5]:10]$
(%i2)   I:matrix([I[1]],[I[2]],[I[3]])$
(%i3)   E:matrix([E[1]+E[6]],[E[2]+E[5]],[E[3]-E[5]+E[6]])$
(%i4)   R:matrix([R[1]+R[4]+R[6],R[4],R[6]],
                 [R[4],R[2]+R[4]+R[5],-R[5]],
                 [R[6],-R[5],R[3]+R[5]+R[6]])$
(%i5)   I=R^^(-1).E;
```

$$(\%o5) \quad \begin{bmatrix} I_1 \\ I_2 \\ I_3 \end{bmatrix} = \begin{bmatrix} 0.03943486760263017 \\ 0.1567442687044606 \\ 0.07938510751732716 \end{bmatrix}$$

```
(%i20)  I1:0.03943486760263017; I2:0.1567442687044606;
        I3:0.07938510751732716;
(I1)    0.03943486760263017
  (I2)    0.1567442687044606
  (I3)    0.07938510751732716
(%i17)  I4:I1+I2; I5:I2-I3; I6:I1+I3;
(I4)    0.1961791363070908
  (I5)    0.07735916118713343
  (I6)    0.1188199751199573
```

IV. 12 Applying Maxwell's principle, derive the ratio of the currents flowing through the parallel connected resistors $R1$ and $R2$ as given in Fig. IV.12.



Fig. IV.12

Solution:

In accordance with *Maxwell*'s principle, the power generated in the circuits is the minimal possible. The power generated across a resistor R, with a current I flowing through may be written as the function

$$P(I,R) = I^2 R$$

The power generated in the whole circuit is the sum of the powers generated across each resistor. The sum of the currents flowing through the resistors gives the resultant current

$$I = I_1 + I_2 \text{ or } I_2 = I - I_1.$$

It can be easily noticed that the last equation, will eliminate variable I_2 from the function describing the resultant power. Considering necessary condition for the minimum value of the function defining the total power $P(I_1{}^2 R_1) + P(I_2{}^2 R_2)$ generated in the circuit (the derivative must be equal to zero), we obtain current I_1.

```
(%i2)   P(I,R):=I^2*R$    I2:I-I1$
(%i3)   solve(diff(P(I1,R1)+P(I2,R2),I1)=0,I1);
                 I R2
(%o3)   [ I1= ------- ]
               R2+R1
(%i4)   I1:(I*R2)/(R2+R1);
          I R2
(I1)    -------
         R2+R1
(%i6)   I2:I-I1$
        ratsimp(I1/I2);
         R2
(%o6)   ----
         R1
```

Using this result we can calculate the required ratio of the currents which turns out to be inversely proportional to the ratio of the resistances as expected.

IV. 13 We consider a battery of N cells. The electromotive force of each of the cells is equal to E and the internal resistance of each cell is R_i. The cells are connected in m branches each of them containing n cells connected in series. Such a system is connected to a resistor of resistance R. How should such a network of cells be arranged in order to give the greatest possible power across the resistor R?

Solution:

We have to find natural numbers n and m. The electromotive force of each row (series connection) is given by

$$E_s = n\, E,$$

and its internal resistance

$$R_s = n\, R_i$$

These rows are connected parallel, thus the equivalent electromotive force of the battery as a whole is equal to

$$E_{bat} = E_s.$$

The resultant internal resistance of the battery is equal to

$$R_{bat} = \frac{R_s}{m}.$$

In accordance with *Ohm*'s law, the current in the circuit is given by

$$I = \frac{E_{bat}}{R_{bat} + R}.$$

The power generated across the external resistor is given by

$$P = I^2\, R.$$

First of all we enter all what we know so far:

```
(%i6)    Es:n*E$   Rs:n*Ri$   Rbat:Rs/m$   Ebat:Es$
         I:Ebat/(Rbat+R)$   P:R*I^2$
```

As we want to find the maximum power we have to calculate the zeros of the first derivative of function P wrt to R.

$N = n\, m$ is the total number of cells.

```
(%i7)    solve([diff(P,R)=0,m*n=N],[n,m]);
```

(%o7) $$[\,[n=-\sqrt{\frac{N\,R}{Ri}},\, m=-\frac{\sqrt{Ri}\,\sqrt{N}}{\sqrt{R}}\,],\,[n=\sqrt{\frac{N\,R}{Ri}},\, m=\frac{\sqrt{Ri}\,\sqrt{N}}{\sqrt{R}}\,]\,]$$

For physical reasons we are only interested in the nonnegative solutions with $n, m \in \mathbb{N}$.

To make sure the extremum we have found is a maximum we should check the sign of the second derivative for the solution obtained.

```
(%i8)   d2:diff(P,R,2);
```

$$(d2)\quad \frac{6\,n^2\,E^2\,R}{\left(R+\frac{n\,Ri}{m}\right)^4}-\frac{4\,n^2\,E^2}{\left(R+\frac{n\,Ri}{m}\right)^3}$$

```
(%i9)   assume(E>0,R>0,Ri>0,N>0)$
(%i10)  sign(subst([n=sqrt((N*R)/Ri),m=(sqrt(Ri)*sqrt(N))/sqrt(R)],d2));
(%o10)  neg
```

%o10 shows that the power obtained from the battery is in fact the maximum power if we take care that *m* and *n* are chosen according to %o7.

IV. 14 We have three identical cells with an electromotive force of E and an internal resistance of r. How should these cells be connected to each other to serve as battery, in order to give the maximum current across an external resistance of R?

Solution:

It is sufficient to consider and to compare the four configurations of connecting the cells to a circuit as illustrated in Fig. IV.14a – d.



Fig. IV.14

The other possible ways of connecting the cells involve different polarization of the cells, and are thus less favourable.

Using *Kirchhoff*'s laws we get the following equations for the four circuits:

(a)

$$E = I_a\left(R+\frac{1}{\frac{1}{r}+\frac{1}{r}+\frac{1}{r}}\right)=I_a\left(R+\frac{r}{3}\right)$$

(b)

$$\begin{cases}2E = I_1(r+r)+I_bR\\ E=(I_b-I_1)r+I_bR\end{cases}$$

(c)

$$\begin{cases}2E = I_1 r+I_c r+I_cR\\ 2E=(I_c-I_1)r+I_cr+I_cR\end{cases}$$

(d)

$$3E = I_d\,(3r+R)$$

We enter the equations or the systems of equations and then solve them for the currents I_a to I_d

```
circuit (a)
(%i8)   eq1:E=I[a]*(R+r/3)$
        solve(eq1,I[a]);
```
$$(\%o8)\quad [\,I_a = \frac{3\,E}{3\,R+r}\,]$$

```
circuit (b)
(%i11)  eq2:2*E=I[1]*2*r+I[b]*R$
        eq3:E=(I[b]-I[1])*r+I[b]*R$
        solve(eliminate([eq2,eq3],[I[1]]),I[b]);
```
$$(\%o11)\quad [\,I_b = \frac{4\,E}{3\,R+2\,r}\,]$$

```
circuit (c)
(%i17)  eq4:2*E=I[1]*r+I[c]*r+I[c]*R$
        eq5:2*E=(I[c]-I[1])*r+I[c]*r+I[c]*R$
        solve(eliminate([eq4,eq5],[I[1]]),I[c]);
```
$$(\%o17)\quad [\,I_c = \frac{4\,E}{2\,R+3\,r}\,]$$

```
circuit (d)
(%i19)  eq6:3*E=I[d]*(3*r+R)$
        solve(eq6,I[d]);
```
$$(\%o19)\quad [\,I_d = \frac{3\,E}{R+3\,r}\,]$$

This was the easy part which could have been done – maybe in less time – without CAS. But the problem is not yet solved.

To continue the comparison procedure of the currents we assign variable names to each of the solutions from above:

```
(%i14)  I[a]:(3*E)/(3*R+r)$
        I[b]:(4*E)/(3*R+2*r)$
        I[c]:(4*E)/(2*R+3*r)$
        I[d]:(3*E)/(R+3*r)$
```

We introduce a positive parameter ε defining the ratio of the resistances $\frac{R}{r}$ to make the comparison of the currents easier.

```
(%i15)  R:ε*r$
```

Circuit (a) is the most favourable if the following inequalities hold:

$$I_a > I_b, I_a > I_c, I_a > I_d \quad \text{i.e. } \frac{I_a}{I_b} > 1, \frac{I_a}{I_c} > 1, \frac{I_a}{I_d} > 1.$$

We solve the system of inequalities given above (loading a special package "`to_poly_solve`" first).

```
(%i16)  load(to_poly_solve)$
(%i17)  %solve(ev([I[a]/I[b]>1,I[a]/I[c]>1,I[a]/I[d]>1,ε>0]),ε);
to_poly_solve: to_poly_solver.mac is obsolete; I'm loading to_poly_solve.mac instead.
```
$$(\%o17)\quad \%\mathrm{union}\left([\,0<\varepsilon,\ \varepsilon<\frac{2}{3}\,]\right)$$

From %o17 we can conclude that circuit (a) is the most favourable for a ratio of resistances in the range $0<\varepsilon<\frac{2}{3}$ or $0<\frac{R}{r}<\frac{2}{3}$ or $R>\frac{2r}{3}$.

The approach for the remaining cases is similar:

```
same for circuit (b)
(%i18) %solve(ev([I[b]/I[a]>1,I[b]/I[c]>1,I[b]/I[d]>1,ε>0]),ε);
(%o18) %union([2/3<ε, ε<1])
same for circuit (c)
(%i19) %solve(ev([I[c]/I[a]>1,I[c]/I[b]>1,I[c]/I[d]>1,ε>0]),ε);
(%o19) %union([1<ε, ε<3/2])
and finally for circuit (d)
(%i20) %solve(ev([I[d]/I[a]>1,I[d]/I[b]>1,I[d]/I[c]>1,ε>0]),ε);
(%o20) %union([3/2<ε])
```

Circuit (b) is the best in the range $\frac{2}{3}<\frac{R}{r}<1$ or $\frac{2r}{3}<R<r$,

Circuit (c) is the best for $1<\frac{R}{r}<\frac{3}{2}$ or $r<R<\frac{3r}{2}$ and circuit (d) is the best for $\frac{R}{r}>\frac{3}{2}$ or $R>\frac{3r}{2}$.

As can easily be seen all intervals are open. We will check now how the circuits behave when the ratio of the resistances is at one of the boundaries of these intervals? What do you expect?

```
(%i21) ev(subst(ε=2/3,[I[a]/I[b],I[a]/I[c],I[a]/I[d]]));
(%o21) [1, 13/12, 11/9]
```

We have $\frac{I_a}{I_b}=1$ which says that the same current flows in both circuits.

```
(%i22) ev(subst(ε=1,[I[c]/I[a],I[c]/I[b],I[c]/I[d]]));
(%o22) [16/15, 1, 16/15]
(%i23) ev(subst(ε=3/2,[I[c]/I[a],I[c]/I[b],I[c]/I[d]]));
(%o23) [11/9, 13/12, 1]
```

For a ratio of 1 the same current flows in circuits (b) and (c) and for the ratio 1.5 the same current flows in circuits (c) and (d).

We could also evaluate the currents for the boundaries, e.g. for the first case:

```
(%i25) R:2*r/3$
       ev([I[a],I[b],I[c],I[d]]);
(%o25) [E/r, E/r, 12 E/(13 r), 9 E/(11 r)]
```

We find the maximum value $1*E/r$ for circuits (a) and (b).

We might ask ourselves how DERIVE and/or TI-NspireCAS will perform solving the system of inequalities? We enter the definitions of the currents (R = rr) and make the try:

$ia:=\frac{3\cdot e}{3\cdot rr+r}:ib:=\frac{4\cdot e}{3\cdot rr+2\cdot r}:ic:=\frac{4\cdot e}{2\cdot rr+3\cdot r}:id:=\frac{3\cdot e}{rr+3\cdot r}$	$\frac{3\cdot e}{3\cdot r+rr}$
$rr:=\varepsilon\cdot r$	$\varepsilon\cdot r$
⚠ $\text{solve}\left(\frac{ia}{ib}>1 \text{ and } \frac{ia}{ic}>1 \text{ and } \frac{ia}{id}>1,\varepsilon\right)\vert\varepsilon>0$	$0<\varepsilon<\frac{2}{3}$
⚠ $\text{solve}\left(\frac{ib}{ia}>1 \text{ and } \frac{ib}{ic}>1 \text{ and } \frac{ib}{id}>1,\varepsilon\right)$	$\frac{2}{3}<\varepsilon<1$
⚠ $\text{solve}\left(\frac{ic}{ia}>1 \text{ and } \frac{ic}{ib}>1 \text{ and } \frac{ic}{id}>1,\varepsilon\right)$	$1<\varepsilon<\frac{3}{2}$
⚠ $\text{solve}\left(\frac{id}{ia}>1 \text{ and } \frac{id}{ib}>1 \text{ and } \frac{id}{ic}>1,\varepsilon\right)$	$\varepsilon<^{-}3 \text{ or } \varepsilon>\frac{3}{2}$
⚠ $\text{solve}\left(\frac{ia}{ib}>1 \text{ and } \frac{ia}{ic}>1 \text{ and } \frac{ia}{id}>1,\varepsilon\right)$	$\frac{^{-}1}{3}<\varepsilon<\frac{2}{3}$

As we can see there is no problem and we don't need any special package or library. The CAS-machine of TI-NspireCAS is – more or less – based on the DERIVE core, so we can be quite sure that DERIVE doesn't have any problems, too.

Entering the inequalities and output of the result are very clear.

In all examples above the currents did not change with time. In the following problems dealing with currents varying over time (unstable currents) will be treated.

IV. 15 The R-C circuit is given by Fig. IV.14. At time zero the capacitor s not charged and switches A and B are open:

Calculate the charge at the capacitor, the potential across the capacitor and the current at time *t* after closing switch A.

Find the potential across the capacitor, as well as the current flowing in the circuit as a function of time after opening switch A and closing switch B.

Present the relationships graphically for $E = 1$, $R = 5$, $C = 4$.



Fig. IV.14

Solution:

First we write the equation resulting from *Kirchhoff*'s 2nd law:

$$E = I\,R + U \quad \text{where} \quad I = \frac{\mathrm{d}Q}{\mathrm{d}t}.$$

The potential across the capacitor is given by the ratio $U = \dfrac{Q}{C}$ which leads to

$$E = R\frac{\mathrm{d}Q}{\mathrm{d}t} + \frac{Q}{C}.$$

After separating the variables and integrating both sides taking into account the initial condition $Q(t{=}0) = 0$ we can write the equation above in integral form:

$$\frac{\mathrm{d}Q}{EC - Q} = \frac{\mathrm{d}t}{RC} \quad \Big| \int$$

$$\int_0^Q \frac{\mathrm{d}Q}{EC - Q} = \int_0^t \frac{\mathrm{d}t}{RC}$$

The solution with respect to *Q* gives the charge depending on time.

```
(%i1)   integrate(1/(E*C-Q),Q,0,Q)=integrate(1/(R*C),t,0,t);
Is Q positive, negative or zero?n;
 Is C E positive, negative or zero?p;
```

$$\text{(\%o1)} \quad \log(C\,E) - \log(C\,E - Q) = \frac{t}{C\,R}$$

```
(%i2)   solve(%,Q);
```

$$\text{(\%o2)} \quad [Q = C\,E\,\%e^{-\frac{t}{C\,R}}\left(\%e^{\frac{t}{C\,R}} - 1\right)]$$

We store this result as our function $Q(t)$ and proceed calculating the potential and current:

```
(%i3)   Q(t):=C*E*%e^(-t/(C*R))*(%e^(t/(C*R))-1);
```

$$\text{(\%o3)} \quad Q(t) := C\,E\,\%e^{\frac{-t}{C\,R}}\left(\%e^{\frac{t}{C\,R}} - 1\right)$$

```
(%i5)  U(t):=Q(t)/C$
       U(t);
```

$$(\%o5)\quad E\,\%e^{-\frac{t}{C\,R}}\left(\%e^{\frac{t}{C\,R}}-1\right)$$

```
(%i7)  I(t):=diff(Q(t),t)$
       I(t);
```

$$(\%o7)\quad \frac{E}{R}-\frac{E\,\%e^{-\frac{t}{C\,R}}\left(\%e^{\frac{t}{C\,R}}-1\right)}{R}$$

```
(%i8)  factor(%);
```

$$(\%o8)\quad \frac{E\,\%e^{-\frac{t}{C\,R}}}{R}$$




```
(%i11) gr1:subst([E=1,R=5,C=4],Q(t))$
       gr2:subst([E=1,R=5,C=4],U(t))$
       gr3:subst([E=1,R=5,C=4],I(t))$
(%i12) plot2d([gr1,gr2,gr3],[t,0,100],[y,0,4],
             [legend,"Q(t)","U(t)","I(t)"]);
```

Finally we plot the graphs in question.

The simpler way to solve the problem is to calculate charge Q by applying the built-in `ode2` function and the initial condition function `ic1`.

```
(%i7)  de:'diff(Q,t)=-Q/(C*R)+E/R;
```

$$(de)\quad \frac{d}{d\,t}\,Q=\frac{E}{R}-\frac{Q}{C\,R}$$

```
(%i8)  qsoln:ode2(de,Q,t);
```

$$(qsoln)\quad Q=\%e^{-\frac{t}{C\,R}}\left(C\,E\,\%e^{\frac{t}{C\,R}}+\%c\right)$$

```
(%i9)  psoln:ic1(qsoln,t=0,Q=0);
```

$$(psoln)\quad Q=\%e^{-\frac{t}{C\,R}}\left(C\,E\,\%e^{\frac{t}{C\,R}}-C\,E\right)$$

The subsequent operations are the same as given above.

```
(%i14) Q:rhs(psoln)$
       [U=expand(Q/C),I=expand(diff(Q,t))];
```

$$(\%o14)\quad [\,U=E-E\,\%e^{-\frac{t}{C\,R}},\ I=\frac{E\,\%e^{-\frac{t}{C\,R}}}{R}\,]$$

What will happen after closing switch B and opening switch A?

The capacitor discharges, thus we have

$$U = E e^{-\frac{1}{RC}},\quad I = -\frac{E e^{-\frac{1}{RC}}}{R}.$$

IV.16 Given is a R-C circuit consisting of time dependent electromotive force of the form $V(t) = V_0 \cos(\omega t)$, a resistor R and a capacitor C, connected in series.

a) Calculate the charge $Q(t)$ when $Q(0) = Q_0$,

b) Plot the graph of $Q(t)$ for $Q_0 = V_0 = R = C = 1$, $\omega = 5$.

Solution:

We have to solve the following differential equation: $R\,Q'(t) + \dfrac{Q(t)}{C} = V_0 \cos(\omega t)$

which can be rewritten in the equivalent form

$$Q'(t) + \frac{Q(t)}{C\,R} = \frac{V_0 \cos(\omega t)}{R}.$$

We use the built-in `ode2`-function together with `ic1` in order to find the particular solution as we did in problem IV.15:

```
(%i1)   de:'diff(Q,t)=-Q/(C*R)+V[0]*cos(ω*t)/R;
```

(de) $$\frac{d}{d\,t}\,Q = \frac{V_0 \cos(t\,\omega)}{R} - \frac{Q}{C\,R}$$

```
(%i2)   Qsoln:ode2(de,Q,t)$
(%i3)   psoln:ic1(Qsoln,t=0,Q=Q[0]);
```

(psoln) $$Q = \frac{V_0\,C^2\,R\,\%e^{\frac{t}{C\,R}}\,\omega \sin(t\,\omega) + V_0\,C\,\%e^{\frac{t}{C\,R}} \cos(t\,\omega) + Q_0\,C^2\,R^2\,\omega^2 - V_0\,C + Q_0}{C^2\,R^2\,\%e^{\frac{t}{C\,R}}\,\omega^2 + \%e^{\frac{t}{C\,R}}}$$

We substitute the data and plot the graph:

```
(%i5)   Q(t):=rhs(subst([Q[0]=1,V[0]=1,C=1, R=1,ω=5],psoln))$
        Q(t);
```

(%o5) $$\frac{\%e^{-t}\,(5\,\%e^{t} \sin(5\,t) + \%e^{t} \cos(5\,t) + 25)}{26}$$

```
(%i6)   plot2d([Q(t)], [t,0,6], [y,-0.2,1],
         [plot_format, gnuplot])$
```



We will solve problem IV.16 using a TI-NspireCAS-toolbox which was produced by Michel Beaudin form Ecole de Technologie Supérieure, Montreal (ETSM), for his students.

The syntax for treating a R-C-circuit is – using the variable names from above – cir_rc(R,C,V,Q_0).



Please compare the results

IV.17 Given is a R-L circuit consisting of time dependent electromotive force of the form $V(t) = V_0\cos(\omega t)$, a resistor R and an inductor L, connected in series.

a) Calculate the charge $Q(t)$ and the current $I(t)$ when $Q(0) = Q_0$ and $I(0) = 0$

b) Plot the graphs of $Q(t)$ and $I(t)$ for $Q_0 = V_0 = R = \omega = 1$, $L = 2$.

Solution:

We have to solve the following differential equation:

$$L\,Q''(t) + R\,Q(t) = V_0\cos(\omega t).$$

```
(%i1)  de:'L*diff(Q(t),t,2)+R*diff(Q(t),t)=V[0]*cos(ω*t);
```

$$\text{(de)}\quad \left(\frac{d}{d\,t}\,Q(t)\right)R+\left(\frac{d^2}{d\,t^2}\,Q(t)\right)L=V_0\cos(t\,\omega)$$

```
(%i2)  Qsoln:ode2(de,Q(t),t);
Is  R  zero or nonzero?n;
```

$$\text{(Qsoln)}\quad Q(t)=\frac{V_0\,R\sin(t\,\omega)-V_0\,L\,\omega\cos(t\,\omega)}{L^2\,\omega^3+R^2\,\omega}+\%k2\,\%e^{-\frac{t\,R}{L}}+\%k1$$

```
-->    Q(t):=(V[0]*R*sin(t*ω)-V[0]*L*ω*cos(t*ω))/
       (L^2*ω^3+R^2*ω)+%k2*%e^(-(t*R)/L)+%k1$
```

It remains to calculate the two constants of integration.

```
(%i5)   eq1:Q(0)=Q[0]$
        eq2:subst(t=0,diff(Q(t),t))=0$
(%i6)   solve([eq1,eq2],[%k1,%k2]);
```

(%o6) $[[\%k1=Q_0, \%k2=\dfrac{V_0\,L}{L^2\,\omega^2+R^2}]]$

```
(%i7)   [%k1:Q[0],%k2:(V[0]*L)/(L^2*ω^2+R^2)]$
```

Now we can define charge $Q(t)$ and $I(t)$ as the derivative of the charge:

```
(%i8)   Q(t);
```

(%o8) $\dfrac{V_0\,R\sin(t\,\omega)-V_0\,L\,\omega\cos(t\,\omega)}{L^2\,\omega^3+R^2\,\omega}+\dfrac{V_0\,L\,\%e^{-\frac{t\,R}{L}}}{L^2\,\omega^2+R^2}+Q_0$

```
(%i9)   I(t):=diff(Q(t),t)$
(%i10)  I(t);
```

(%o10) $\dfrac{V_0\,L\,\omega^2\sin(t\,\omega)+V_0\,R\,\omega\cos(t\,\omega)}{L^2\,\omega^3+R^2\,\omega}-\dfrac{V_0\,R\,\%e^{-\frac{t\,R}{L}}}{L^2\,\omega^2+R^2}$

We calculate charge and current for the given data

```
(%i11)  example:subst([Q[0]=1,V[0]=1,R=1,L=2,ω=1],[Q(t),I(t)]);
```

(example) $[\dfrac{\sin(t)-2\cos(t)}{5}+\dfrac{2\,\%e^{-\frac{t}{2}}}{5}+1, \dfrac{2\sin(t)+\cos(t)}{5}-\dfrac{\%e^{-\frac{t}{2}}}{5}]$

Finally we prepare for plotting and then we plot the requested functions:

```
(%i13)  Q1(t):=(sin(t)-2*cos(t))/5+(2*%e^(-t/2))/5+1$
        I1(t):=(2*sin(t)+cos(t))/5-%e^(-t/2)/5$
-->     plot2d([Q1(t),I1(t)],[t,0,20],[y,-1,2],
            [legend,"Charge","Current"]);
```



Let's try Michel Beaudin's toolbox once more. There is also a function provided for treating an R-L-circuit:

$kit\_ets\_mb\backslash cir\_rl(1,2,\cos(t),0)$ $\quad \dfrac{-e^{\frac{-t}{2}}}{5}+\dfrac{\cos(t)}{5}+\dfrac{2\cdot\sin(t)}{5}$

$current(t):=\dfrac{-e^{\frac{-t}{2}}}{5}+\dfrac{\cos(t)}{5}+\dfrac{2\cdot\sin(t)}{5}$ $\quad$ *Done*

$q(t):=\int current(t)\,dt+c$ $\quad$ *Done*

$q(0)=1$ $\quad c=1$

$charge(t):=q(t)|c=1$ $\quad$ *Done*

$charge(t)$ $\quad \dfrac{2\cdot e^{\frac{-t}{2}}}{5}-\dfrac{2\cdot\cos(t)}{5}+\dfrac{\sin(t)}{5}+1$



IV.18 The R-L-C circuit illustrated in Fig. IV.18 is composed of a coil of resistance R and inductance L, together with a capacitor of capacitance C. The charge accumulated on the capacitor is Q_0. The capacitor discharges after closing switch K. Find the charge on the capacitor as a function of time.



Fig. IV. 18

Solution:

We have to solve the following differential equation:

$$L\frac{\mathrm{d}I}{\mathrm{d}t}+RI+\frac{Q}{C}=0.$$

where $I=\dfrac{dQ}{dt}$ denotes the current. Thus, the charge on the capacitor is given by the differential equation

$$\frac{\mathrm{d}^2Q}{\mathrm{d}t^2}+\frac{R}{L}\frac{\mathrm{d}Q}{\mathrm{d}t}+\frac{Q}{LC}=0 \leftrightarrow Q''+\frac{R}{L}Q'+\frac{1}{LC}Q=0.$$

```
(%i1)    de:'diff(Q(t),t,2)+R/L*diff(Q(t),t)+Q(t)/(L*C)=0;
```

(de) $$\frac{\left(\frac{d}{d\,t}\,Q(t)\right)R}{L}+\frac{Q(t)}{C\,L}+\frac{d^2}{d\,t^2}\,Q(t)=0$$

We try to solve this equation by applying function `ode2`:

```
(%i2)    Qsoln:ode2(de,Q(t),t);
Is C(C R^2-4 L) positive, negative or zero?n;
```

(Qsoln) $$Q(t)=\%e^{-\frac{t\,R}{2\,L}}\left(\%k1\,\sin\left(\frac{t\sqrt{\frac{4}{C\,L}-\frac{R^2}{L^2}}}{2}\right)+\%k2\,\cos\left(\frac{t\sqrt{\frac{4}{C\,L}-\frac{R^2}{L^2}}}{2}\right)\right)$$

The above solution is correct for $C\,(C\,R^2-4L)<0$.

In the next step we evaluate the integration constants by applying function `ic2`:

```
(%i3)    psoln:ic2(Qsoln,t=0,Q(t)=Qo,diff(Q(t),t)=0);
```

(psoln) $$Q(t)=\%e^{-\frac{t\,R}{2\,L}}\left(\frac{Q(0)\,R\,\sin\left(\frac{t\sqrt{\frac{4}{C\,L}-\frac{R^2}{L^2}}}{2}\right)}{L\sqrt{\frac{4}{C\,L}-\frac{R^2}{L^2}}}+Q(0)\,\cos\left(\frac{t\sqrt{\frac{4}{C\,L}-\frac{R^2}{L^2}}}{2}\right)\right)$$

The particular solution $Q(t)$ can be rewritten in simpler form. Namely replacing the sub-expression $\frac{1}{2}\sqrt{\frac{4}{LC}-\frac{R^2}{L^2}}$ with ω which exhibits damped oscillations (numerical data: $Q_0 = R = L = 1$, $\omega = 5$).

(If $\omega = 5$ then $C = 4/101$.)

```
(%i4)  Q(t):=%e^(-(t*R)/(2*L))*((Q0*R*sin((t*sqrt(4/(C*L)-R^2/L^2))/2))/
       (L*sqrt(4/(C*L)-R^2/L^2))+Q0*cos((t*sqrt(4/(C*L)-R^2/L^2))/2))$
(%i5)  Q1(t):=%e^((-t*R)/(2*L))*Q0*(R*sin(ω*t)/(2*ω*L)+cos(ω*t));
```

(%o5) $Q1(t) := \%e^{\frac{-t\,R}{2\,L}}\, Q0 \left(\frac{R\sin(\omega\,t)}{2\,\omega\,L}+\cos(\omega\,t)\right)$

```
(%i7)  [Q0:1,R:1,L:1,C:4/101,ω:5]$
       [Q1(t),Q(t)];
```

(%o7) $\left[\%e^{-\frac{t}{2}}\left(\frac{\sin(5\,t)}{10}+\cos(5\,t)\right), \%e^{-\frac{t}{2}}\left(\frac{\sin(5\,t)}{10}+\cos(5\,t)\right)\right]$

```
(%i8)  plot2d([Q(t)],[t,0,10],[y,-1,1])$
```



We would like to present solution curves $Q(t)$ with various values for ω. For this purpose we define function Q_ and plot its graph with $\omega = 2$, 5, 10 and 20.

We shift each function graph by 2 in y-direction.

```
(%i9)  Q_(ω_,t):=%e^((-t*R)/(2*L))*Q0*(R*sin(ω_*t)/(2*ω_*L)+cos(ω_*t));
```

(%o9) $Q\_(\omega\_, t) := \%e^{\frac{-t\,R}{2\,L}}\, Q0 \left(\frac{R\sin(\omega\_\,t)}{2\,\omega\_\,L}+\cos(\omega\_\,t)\right)$

```
(%i10) [Q1_:Q_(2,t),Q2_:Q_(5,t),Q3_:Q_(10,t),Q4_:Q_(20,t)]$
```

```
-->     wxdraw2d(xaxis=true, yaxis=true, xrange=[0,6], yrange=[-1,9],
                  color = black, key="omega=2",explicit(Q_(2,t),t,0,6),
                  color = red, key="omega=5",explicit(Q_(5,t)+2,t,0,6),
                  color = blue, key="omega=10",explicit(Q_(10,t)+4,t,0,6),
                  color = green, key="omega=20",explicit(Q_(20,t)+6,t,0,6))$
```

(%t12)



If the parameters of the problem satisfy the inequality $C\,(C\,R^2 - 4L) > 0$ we get in a similar way:

```
(%i1)    de:'diff(Q(t),t,2)+R/L*diff(Q(t),t)+Q(t)/(L*C)=0$
(%i2)    Qsoln:ode2(de,Q(t),t);
Is  C(C R^2 -4 L)  positive, negative or zero?p;
```

(Qsoln) $$Q(t)=\%k1\,\%e^{\frac{t\left(\sqrt{\frac{R^2}{L^2}-\frac{4}{C\,L}}-\frac{R}{L}\right)}{2}}+\%k2\,\%e^{\frac{t\left(-\sqrt{\frac{R^2}{L^2}-\frac{4}{C\,L}}-\frac{R}{L}\right)}{2}}$$

```
(%i3)    psoln:ic2(Qsoln,t=0,Q(t)=Q[0],diff(Q(t),t)=0)$
(%i4)    subst([Q(0)=1,C=1,R=3,L=1],psoln);
```

(%o4) $$Q(t)=\frac{(\sqrt{5}+3)\,\%e^{\frac{(\sqrt{5}-3)\,t}{2}}}{2\sqrt{5}}+\frac{(\sqrt{5}-3)\,\%e^{\frac{(-\sqrt{5}-3)\,t}{2}}}{2\sqrt{5}}$$

```
(%i5)    Q(t):=((sqrt(5)+3)*%e^(((sqrt(5)-3)*t)/2))/(2*sqrt(5))
              +((sqrt(5)-3)*%e^(((-sqrt(5)-3)*t)/2))/(2*sqrt(5))$
(%i6)    plot2d([Q(t)],[t,0,10],[y,0,1])$
```

It can easily be seen that in this case charge $Q(t)$ exponentially decreases with time. The graphical representation is following.



How to do with DERIVE and TI-NspireCAS?

We solve this equation by applying the DSOLVE2_IV function:

$$\#3: \quad Q(t) := \text{DSOLVE2\_IV}\left(\frac{R}{L}, \frac{1}{L \cdot C}, 0, t, 0, Q0, 0\right)$$

Simplifying #3, we get very bulky result in which the functions SIGN and ABS appear. Therefore we omit displaying it. By declaring the appropriate domain for the variable L We get the solution in a slightly simpler form (we also omit displaying it) with subexpressions of the form $\sqrt{(CR^2 - 4L)/C}$.

$$\#5: \quad L :\in \text{Real } (0, \infty)$$

Therefore, further simplification of Q(t) can be carried out, depending upon the values of the expressions under the square roots.

Suppose the parameters of the exercise satisfy the inequality $\dfrac{CR^2 - 4L}{C} > 0$. Then replacing $\dfrac{CR^2 - 4L}{C}$ by $-\omega^2$ in function $Q(t)$ gives a much simpler expression

$$\#7: \quad Q1(t) := \text{SUBST}\left(Q(t), \frac{C \cdot R^2 - 4 \cdot L}{C}, -\omega^2\right)$$

$$\#8: \quad Q1(t) := e^{-R \cdot t/(2 \cdot L)} \cdot \left(Q0 \cdot \text{COS}\left(\frac{t \cdot \omega}{2 \cdot L}\right) + \frac{Q0 \cdot R \cdot \text{SIN}\left(\frac{t \cdot \omega}{2 \cdot L}\right)}{\omega}\right)$$

which exhibits damped oscillations.

For Q0=1, C=4/101, R=1, L=1 and $\omega = 10$ we get

$$\#9: \quad \left[Q0 := 1,\ C := \frac{4}{101},\ R := 1,\ \omega := 10,\ L := 1\right]$$

$$\#10: \quad [Q1(t),\ Q(t)] = \left[e^{-t/2}\cdot\left(\mathrm{COS}(5\cdot t) + \frac{\mathrm{SIN}(5\cdot t)}{10}\right),\ e^{-t/2}\cdot\left(\mathrm{COS}(5\cdot t) + \frac{\mathrm{SIN}(5\cdot t)}{10}\right)\right]$$

which is identical with the Maxima-solution.

In a similar way we get the solution for the other case:

$$\#11: \quad e^{t\cdot(\sqrt{5}/2 - 3/2)}\cdot\left(\frac{3\cdot\sqrt{5}}{10} + \frac{1}{2}\right) + e^{-t\cdot(\sqrt{5}/2 + 3/2)}\cdot\left(\frac{1}{2} - \frac{3\cdot\sqrt{5}}{10}\right)$$

We don't show the graphs. Finally we will try solving the problem with TI-NspireCAS:

$$\mathrm{deSolve}\left(q'' + \frac{r}{l}\cdot q' + \frac{q}{l\cdot c} = 0 \text{ and } q(0) = q0 \text{ and } q'(0) = 0, t, q\right)$$

$$q = \frac{q0\cdot\left(l\cdot\sqrt{\frac{c\cdot r^2 - 4\cdot l}{c}} + |l|\cdot r\right)\cdot e^{\frac{\sqrt{\frac{c\cdot r^2-4\cdot l}{c}}\cdot t}{2\cdot|l|} - \frac{r\cdot t}{2\cdot l}}}{2\cdot l\cdot\sqrt{\frac{c\cdot r^2-4\cdot l}{c}}} + \frac{q0\cdot\left(l\cdot\sqrt{\frac{c\cdot r^2 - 4\cdot l}{c}} - |l|\cdot r\right)\cdot e^{\frac{-\sqrt{\frac{c\cdot r^2-4\cdot l}{c}}\cdot t}{2\cdot|l|} - \cdots}}{2\cdot l\cdot\sqrt{\frac{c\cdot r^2-4\cdot l}{c}}}$$

$$q(t) := \frac{q0\cdot\left(l\cdot\sqrt{\frac{c\cdot r^2 - 4\cdot l}{c}} + |l|\cdot r\right)\cdot e^{\frac{\sqrt{\frac{c\cdot r^2-4\cdot l}{c}}\cdot t}{2\cdot|l|} - \frac{r\cdot t}{2\cdot l}}}{2\cdot l\cdot\sqrt{\frac{c\cdot r^2-4\cdot l}{c}}} + \frac{q0\cdot\left(l\cdot\sqrt{\frac{c\cdot r^2 - 4\cdot l}{c}} - |l|\cdot r\right)\cdot e^{\frac{-\sqrt{\frac{c\cdot r^2-4\cdot l}{c}}\cdot t}{2\cdot|l|} - \cdots}}{2\cdot l\cdot\sqrt{\frac{c\cdot r^2-4\cdot l}{c}}}$$

Done

$q1(t) := q(t)|q0=1 \text{ and } r=1 \text{ and } l=1 \text{ and } c=\frac{4}{101}$ Done

$q2(t) := q(t)|q0=1 \text{ and } r=3 \text{ and } l=1 \text{ and } c=1$ Done

$$q1(t) \qquad \frac{e^{\frac{-t}{2}}\cdot(10\cdot\cos(5\cdot t)+\sin(5\cdot t))}{10}$$

$$q2(t) \qquad \left(\frac{3\cdot\sqrt{5}}{10}+\frac{1}{2}\right)\cdot e^{\left(\frac{\sqrt{5}}{2}-\frac{3}{2}\right)\cdot t} + \left(\frac{1}{2}-\frac{3\cdot\sqrt{5}}{10}\right)\cdot e^{\left(\frac{-\sqrt{5}}{2}-\frac{3}{2}\right)\cdot t}$$

$f1(x) := q1(x) : f2(x) := q2(x)$ Done

$$f1(x) := \frac{e^{\frac{-x}{2}}\cdot(10\cdot\cos(5\cdot x)+\sin(5\cdot x))}{10}$$ Done

$$f2(x) := \left(\frac{3\cdot\sqrt{5}}{10}+\frac{1}{2}\right)\cdot e^{\left(\frac{\sqrt{5}}{2}-\frac{3}{2}\right)\cdot x} + \left(\frac{1}{2}-\frac{3\cdot\sqrt{5}}{10}\right)\cdot e^{\left(\frac{-\sqrt{5}}{2}-\frac{3}{2}\right)\cdot x}$$ Done

It is not surprising that the solution appearing is also a very bulky expression. But it needs only little work to obtain the solutions for our data from above.



Nice graphs are in return for our efforts.

The following examples of this part are more complicated.

IV.19 Fig. IV.19 shows a circuit. Calculate the frequency and the form of the oscillations in this circuit.



Fig. IV.19

Solution:

We postulate that the dependence between the charge on the capacitors and time is harmonic as $Q_k = A_k e^{i\omega t}, k = 1,2,3.$

We enter these dependencies together with the relations resulting from the application of *Kirchhoff*'s laws, first for each junction and then for each circuit:

```
(%i1)  Q[k](t):=A[k]*%e^(%i*ω*t);
```

$$\text{(\%o1)}\quad Q_k(t) := A_k\, \%e^{\%i\,\omega\, t}$$

– for each junction

```
(%i4)  I[1](t):=I[2](t)+diff(Q[1](t),t);
       I[2](t):=diff(Q[2](t),t);
       I[3](t):=I[2](t)-diff(Q[3](t),t);
```

$$\text{(\%o2)}\quad I_1(t) := I_2(t) + \mathrm{diff}(Q_1(t), t)$$

$$\text{(\%o3)}\quad I_2(t) := \mathrm{diff}(Q_2(t), t)$$

$$\text{(\%o4)}\quad I_3(t) := I_2(t) - \mathrm{diff}(Q_3(t), t)$$

– for each circuit

```
(%i7)  eq1:L*diff(I[1](t),t)+Q[1](t)/C=0;
       eq2:L*diff(I[3](t),t)-Q[3](t)/C=0;
       eq3:Q[2](t)/C[0]+Q[3](t)/C-Q[1](t)/C=0;
```

$$\text{(eq1)}\quad L\left(-A_2\,\omega^2\,\%e^{\%i\,t\,\omega} - A_1\,\omega^2\,\%e^{\%i\,t\,\omega}\right) + \frac{A_1\,\%e^{\%i\,t\,\omega}}{C} = 0$$

$$\text{(eq2)}\quad L\left(A_3\,\omega^2\,\%e^{\%i\,t\,\omega} - A_2\,\omega^2\,\%e^{\%i\,t\,\omega}\right) - \frac{A_3\,\%e^{\%i\,t\,\omega}}{C} = 0$$

$$\text{(eq3)}\quad \frac{A_3\,\%e^{\%i\,t\,\omega}}{C} - \frac{A_1\,\%e^{\%i\,t\,\omega}}{C} + \frac{A_2\,\%e^{\%i\,t\,\omega}}{C_0} = 0$$

Factor $e^{it\omega}$ differs from zero, thus we can rewrite the equations in the following simpler form:

```
(%i9)  eqs:[eq1,eq2,eq3]$
       eqs:expand((eqs/%e^(%i*ω*t)));
```

$$\text{(eqs)}\quad \left[-A_2\,L\,\omega^2 - A_1\,L\,\omega^2 + \frac{A_1}{C} = 0,\ A_3\,L\,\omega^2 - A_2\,L\,\omega^2 - \frac{A_3}{C} = 0,\ \frac{A_3}{C} - \frac{A_1}{C} + \frac{A_2}{C_0} = 0\right]$$

We obtain a non-trivial solution of a homogeneous system of linear equations, if the determinant of the matrix of coefficients is equal to zero. Applying this condition leads to the following matrix:

```
(%i10)  m[i,j]:=coeff(lhs(eqs[i]),A[j]);
(%o10)  m_{i,j}:=coeff(lhs(eqs_i),A_j)
(%i11)  M:genmatrix(m,3,3);
```

$$
(M)\quad \begin{bmatrix} \frac{1}{C}-L\omega^2 & -L\omega^2 & 0 \\ 0 & -L\omega^2 & L\omega^2-\frac{1}{C} \\ -\frac{1}{C} & \frac{1}{C_0} & \frac{1}{C} \end{bmatrix}
$$

The solution with respect to ω gives the required frequencies in the circuit. Because of physical reasons we are only interested in the positive solutions.

```
(%i12)  solve(determinant(M)=0,ω);
```

$$
(\%o12)\quad [\omega=-\frac{1}{\sqrt{C\,L+2\,C_0\,L}},\ \omega=\frac{1}{\sqrt{C\,L+2\,C_0\,L}},\ \omega=-\sqrt{\frac{1}{C\,L}},\ \omega=\sqrt{\frac{1}{C\,L}}]
$$

```
(%i13)  [ω1:sqrt(1/(C*L)),ω2:1/sqrt(C*L+2*C[0]*L)]$
```

$$
\omega_1=\frac{1}{\sqrt{LC}},\quad \omega_2=\frac{1}{\sqrt{L(C+2C_0)}}
$$

In order to calculate the amplitudes A_k it is necessary to solve the system of equations `eqs` (%i9) for both frequencies. We start with ω_1:

```
(%i14)  solve(subst(ω=ω1,eqs),[A[1],A[2],A[3]]);
solve: dependent equations eliminated: (2)
```

$$
(\%o14)\quad [[A_1=\%r1,\ A_2=0,\ A_3=\%r1]]
$$

Solution %o14 indicates that there is no change of charge at the capacitor C_0 ($A_2 = 0$) whereas the charges at the other capacitors are equal ($A_1 = A_3$).

Proceeding in a similar way for the next frequency we get:

```
(%i15)  solve(subst(ω=ω2,eqs),[A[1],A[2],A[3]]);
solve: dependent equations eliminated: (3)
```

$$
(\%o15)\quad [[A_1=\frac{C_0}{2},\ A_2=\%r2,\ A_3=-\frac{C_0}{2}]]
$$

The charges at Capacitors C_1 and C_3 have opposite phases ($A_1 = -A_3$).

Resuming the solutions (the amplitudes) can be rewritten in the form:

- for $\omega = \omega_1$: $A_1 = A_3 = const1$, $A_2 = 0$
- for $\omega = \omega_2$: $B_1=\frac{C_0}{2}$, $B_2 = const2$, $B_3=-\frac{C_0}{2}$

```
(%i18)  A[1]:const1$  A[2]:0$   A[3]:const1$
(%i21)  B[1]:C[0]/2$  B[2]:const2$   B[3]:-B[1]$
```

The charges on the capacitors are a linear combination of these solutions:

```
(%i22)  Q(k,t):=A[k]*%e^(%i*ω1*t)+B[k]*%e^(%i*ω2*t);
```

(%o22) $Q(k,t):=A_k\,\%e^{\%i\,\omega 1\,t}+B_k\,\%e^{\%i\,\omega 2\,t}$

In the final step we calculate the real values of the charges on the capacitors.

```
(%i23)  assume(L>0,C>0,t>=0,C[0]>0);
```

(%o23) $[L>0, C>0, t>=0, C_0>0]$

```
(%i24)  realpart(Q(1,t));
```

(%o24) $$\frac{C_0\cos\left(\frac{t}{\sqrt{C\,L+2\,C_0\,L}}\right)}{2}+const1\cos\left(t\sqrt{\frac{1}{C\,L}}\right)$$

```
(%i25)  realpart(Q(2,t));
```

(%o25) $$const2\cos\left(\frac{t}{\sqrt{C\,L+2\,C_0\,L}}\right)$$

```
(%i26)  realpart(Q(3,t));
```

(%o26) $$const1\cos\left(t\sqrt{\frac{1}{C\,L}}\right)-\frac{C_0\cos\left(\frac{t}{\sqrt{C\,L+2\,C_0\,L}}\right)}{2}$$

Constants *const1* and *const2* can be found from the initial conditions.

As an exercise we will analyse the obtained results for two cases:

First case is for $C_0 = 0$ (no coupling):

```
(%i29)  Q1(t):=realpart(Q(1,t))$  Q2(t):=realpart(Q(2,t))$
        Q3(t):=realpart(Q(3,t))$
(%i36)  Q1_(t):=subst([C[0]=0, C=5*10^(-6), L=10^(-3),
                        const1=1],Q1(t))$
        Q2_(t):=subst([C[0]=0, C=5*10^(-6), L=10^(-3),
                        const2=2],Q2(t))$
        Q3_(t):=subst([C[0]=0, C=5*10^(-6), L=10^(-3),
                        const1=1],Q3(t))$
(%i37)  draw2d(title="Co = 0",
               xaxis = true, yaxis = true,
               xrange=[0,0.001],yrange=[-2.01,2.01],
               color = black, key="Q(1,t)",
                  explicit(Q1_(t),t,0,0.001),
               color = red, key="Q(2,t)",
                  explicit(Q2_(t),t,0,0.001),
               color = blue, key="Q(3,t)",
                  explicit(Q3_(t),t,0,0.001));
(%o37)  [gr2d(explicit,explicit,explicit)]
```



Graphs of $Q(1,t)$ and $Q(3,t)$ are identical.

Second case is for = $C_0 >> C$, $C_0 = 10^{-4}$:

```
(%i37)  Q1_(t):=subst([C[0]=10^(-4), C=5*10^(-6), L=10^(-3),
                   const1=1],Q1(t))$
        Q2_(t):=subst([C[0]=10^(-4), C=5*10^(-6), L=10^(-3),
                   const2=2],Q2(t))$
        Q3_(t):=subst([C[0]=10^(-4), C=5*10^(-6), L=10^(-3),
                   const1=1],Q3(t))$
(%i41)  draw2d(title="Co=10^-4, C0>>C",
               xaxis = true, yaxis = true,
               xrange=[0,0.005],yrange=[-2.01,2.01],
               color = black, key="Q(1,t)",
                  explicit(Q1_(t)+0.1,t,0,0.005),
               color = red, key="Q(2,t)",
                  explicit(Q2_(t),t,0,0.005),
               color = blue, key="Q(3,t)",
                  explicit(Q3_(t),t,0,0.005));
```



IV.20 Show that one can choose the resistors R_1 and R_2 in the circuit given in Fig. IV.20 in such a way that the impedance of the circuit is real regardless of the frequency ω.



Fig. IV.20

Solution:

The impedance of the circuit as a whole is given by the relationship

$$Z = \frac{1}{\frac{1}{R_1}+\frac{1}{R_C}} + \frac{1}{\frac{1}{R_2}+\frac{1}{R_L}}$$

where $R_C = \dfrac{1}{i\omega C}$ denotes the impedance of the capacitor and $R_L = i\omega L$ denotes the impedance of the solenoid.

We start defining the variables and declaring the domains.

```
(%i3)   R[C]:1/(%i*ω*C)$   R[L]:%i*ω*L$
        Z:1/1/(1/R[1]+1/R[C])+1/((1/R[2]+1/R[L]))$
(%i4)   assume(R[1]>0, R[2]>0,L>0,C>0,ω>0)$
```

Impedance Z is real if its imaginary part equals zero. Thus, we obtain the following equation which we then solve with respect to, e.g. R_1.

```
(%i5)   eq:imagpart(Z)=0;
```

(eq) $$\frac{1}{L\left(\frac{1}{L^2\,\omega^2}+\frac{1}{R_2^2}\right)\omega} - \frac{C\,\omega}{C^2\,\omega^2+\frac{1}{R_1^2}} = 0$$

```
(%i6)   solve(eq,R[1]);
```

(%o6) $$\left[R_1 = -\frac{R_2\sqrt{L}}{\sqrt{C}\sqrt{L^2\,\omega^2-R_2^2\,C\,L\,\omega^2+R_2^2}},\ R_1 = \frac{R_2\sqrt{L}}{\sqrt{C}\sqrt{L^2\,\omega^2-R_2^2\,C\,L\,\omega^2+R_2^2}}\right]$$

We are interested in the positive solution only.

```
(%i7)   R[1]:rhs(%[2]);
```

(%o7) $$\frac{R_2\sqrt{L}}{\sqrt{C}\sqrt{L^2\,\omega^2-R_2^2\,C\,L\,\omega^2+R_2^2}}$$

The above relationship between R_1 and R_2 is the result of the condition that the impedance is real.

We know also that both resistances don't depend on the frequency ω. Hence the equation $\dfrac{dR_1}{d\omega}=0$ must hold.

Its positive solution with respect to R_2 gives:

```
(%i8)   solve(diff(R[1],ω)=0,R[2]);
(%o8)   [R_2=-√L/√C, R_2=√L/√C, R_2=0]
(%i9)   R[2]:sqrt(L)/sqrt(C)$
```

Finally we calculate R_1 and verify the solution:

```
(%i10)  ev(R[1]);
(%o10)  √L/√C
(%i11)  imagpart(z);
(%o11)  0
```

It has been shown that for resistances $R_1 = R_2 = \sqrt{\dfrac{L}{C}}$ the impedance of the circuit is real for all frequencies.

IV.21 A constant potential is connected to a bridge circuit (Fig. IV.21). The bridge was braught to equilibrium by changing resistance R_2.In equilibrium state the resistances observed were equal to R_1, R_2 and R_3. Then a sinusoidal variable potential was connected to the bridge and the circuit was brought back into equilibrium by just changing the capacity C. Given R_1, R_2, R_3 and C calculate the resistance R of the coil and its inductance L.



Fig. IV.21

Solution:

Equilibrating the bridge means that there is no current flowing through the galvanometer. Thus, the equality of the appropriate drops in potential across the arms of the circuit leads to the equations:

– for the alternating current

$$I_1(R+i\omega L+R_3)=I_2\left(R_1+\frac{1}{\frac{1}{R_2}}+i\omega C\right) \text{ and } I_1(R+i\omega L)=I_2\,R_1$$

where I_1 the current flowing through the coil and I_2 the current flowing through the resistor R_1.

– for the direct current

$$R_1 R_3 = R_2 R.$$

```
(%i1) eq1:R1*(R+%i*omega*L+R3)=I2*(R1+1/(1/R2+%i*omega*C));
      eq2:I1*(R+%i*omega*L)=I2*R1;
      eq3:R1*R3=R2*R;
```

(%o1) $R1\,(R3+R+\%i\,\omega\,L)=I2\left(\dfrac{1}{\dfrac{1}{R2}+\%i\,\omega\,C}+R1\right)$

(%o2) $I1\,(R+\%i\,\omega\,L)=I2\;R1$

(%o3) $R1\;R3=R\;R2$

```
(%i4) solve(eliminate([eq1,eq2,eq3],[I1]),[R,L]);
```

(%o4) $[\,[R=\dfrac{R1\;R3}{R2},\;L=$

$$\dfrac{(\%i\,\omega\,C\,R1\,R2^2+(\%i\,\omega\,C\,R1^2+R1)\,R2+R1^2)\,R3+(-\%i\,\omega\,C\,I2\,R1-I2)\,R2^2-I2\,R1\,R2}{\omega^2\,C\,R1\,R2^2-\%i\,\omega\,R1\,R2}]$$

IV.22 We are given an infinite system consisting of capacitors C and inductances L (Fig. IV.22). The potential U applied to this system varies according to the formula $U = U_0 \cos(\omega t)$. Calculate the effective current at the voltage input.



Fig. IV.22

Solution:

We use the result obtained in problem V.6, when we calculated the resultant resistance of an infinite system of resistors. Comparing Fig. IV.6a and Fig. IV.22 it can easily be seen tat the role of R_1 is taken by the impedance of the inductance Z_L, and the role of the resistance R_2 is taken by the impedance of the capacitor Z_C. Thus in accordance with formula %o3 from this problem the impedance of the system as a whole is given by

$$Z=\frac{\sqrt{Z_L^{\,2}+4Z_L Z_C}+Z_L}{2}\quad\text{with}\quad Z_L=i\omega L,\ Z_C=\frac{1}{i\omega C}.$$

The effective current is given by $I_{eff}=\dfrac{U_{eff}}{\sqrt{Z\,Z^*}}$ where $U_{eff}=\sqrt{\dfrac{1}{T}\int_0^T U^2\mathrm{d}t}$ is the potential and the sign "*" denotes the complex conjugate.

We enter the formulae:

```
(%i4)    Z[L]:%i*ω*L$  Z[C]:1/(%i*ω*C)$
         Z:(sqrt(Z[L]^2+4*Z[L]*Z[C])+Z[L])/2$
         U:U[0]*cos(ω*t)$
(%i6)    T:2*%pi/ω$
         U[eff]:sqrt(1/T*integrate(U^2,t,0,T))$
Is ω positive or negative?p;
```

Then we try to evaluate the effective current I_{eff}:

```
(%i8)    assume(L>0,C>0,ω>0,U[0]>0)$

         I[eff]:U[eff]/cabs(Z);
```

(%o8) $$(\sqrt{2}\left|U_0\right|)/\mathrm{sqrt}\left(\cos\left(\frac{\mathrm{atan2}\left(0,\frac{4L}{C}-L^2\omega^2\right)}{2}\right)^2\left|L^2\omega^2-\frac{4L}{C}\right|+\left(\sin\left(\frac{\mathrm{atan2}\left(0,\frac{4L}{C}-L^2\omega^2\right)}{2}\right)\sqrt{\left|L^2\omega^2-\frac{4L}{C}\right|}+L\omega\right)^2\right)$$

`(atan2(y,x)` yields the value of atan(y/x) in the interval -%pi to %pi. atan2 is equal zero, so we can simplify the above result by substitution

```
(%i9)    I[eff]:ratsimp(subst(atan2(0,(4*L)/C-L^2*ω^2)=0,I[eff]));
```

(%o9) $$\frac{\sqrt{2}U_0\sqrt{C}}{\sqrt{\left|CL^2\omega^2-4L\right|+CL^2\omega^2}}$$

Now we have to consider two cases:

For small frequencies one can simplify the function `abs` and for large frequencies the `abs` (in expression `Ieff`) can be omitted.

```
(%i10)   (sqrt(2)*U[0]*sqrt(C))/sqrt(-(C*L^2*ω^2-4*L)+C*L^2*ω^2);
```

(%o10) $$\frac{U_0\sqrt{C}}{\sqrt{2}\sqrt{L}}$$

```
(%i11)   (sqrt(2)*U[0]*sqrt(C))/sqrt((C*L^2*ω^2-4*L)+C*L^2*ω^2);
```

(%o11) $$\frac{\sqrt{2}U_0\sqrt{C}}{\sqrt{2CL^2\omega^2-4L}}$$

It can easily be noticed that this expression is a decreasing function on frequency ω.

Resuming the results `%o12` and `%o13` can be written together in the concise form of a piecewise defined function.

```
(%i12) Ieff(ω):=if ω<=2/sqrt(L*C) then
                (U[0]*sqrt(C))/(sqrt(2)*sqrt(L))
                else
                (sqrt(2)*U[0]*sqrt(C))/sqrt(2*C*L^2*ω^2-4*L);
```

(%o12) $\mathrm{Ieff}(\omega):=\mathrm{if}\ \omega<=\frac{2}{\sqrt{L\,C}}\ \mathrm{then}\ \frac{U_0\sqrt{C}}{\sqrt{2}\sqrt{L}}\ \mathrm{else}\ \frac{\sqrt{2}\,U_0\sqrt{C}}{\sqrt{2\,C\,L^2\,\omega^2-4\,L}}$

The system analysed is thus a *low-pass filter*.

We plot I_{eff} for $U0 = 1$, $L = 1$ and $C = 1$ and get

```
(%i13) plot2d(subst([U[0]=1,L=1,C=1],Ieff(omega)),[omega,0,10]);
```



Follow the procedure performed with TI-NspireCAS:

$z1:=i\cdot\omega\cdot l:z2:=\frac{1}{i\cdot\omega\cdot c}$	$\frac{-1}{c\cdot\omega}\cdot i$
$z:=\frac{\sqrt{z1^2+4\cdot z1\cdot z2}+z1}{2}$	$\frac{\sqrt{\frac{-(c\cdot l\cdot\omega^2-4)\cdot l}{c}}+l\cdot\omega\cdot i}{2}$
$u:=u0\cdot\cos(\omega\cdot t)$	$u0\cdot\cos(\omega\cdot t)$
$tt:=\frac{2\cdot\pi}{\omega}$	$\frac{2\cdot\pi}{\omega}$
$u\_eff:=\sqrt{\frac{1}{tt}\cdot\int_0^{tt}u^2\,dt}$	$\frac{\lvert u0\rvert\cdot\sqrt{2}}{2}$
$z\_:=\frac{\sqrt{\frac{-a\cdot l}{c}}+l\cdot\omega\cdot i}{2}\,\Big\vert\, a<0$	$\frac{\sqrt{\frac{-a\cdot l}{c}}+l\cdot\omega\cdot i}{2}$
$\frac{u\_eff}{\sqrt{(\mathrm{real}(z\_))^2+(\mathrm{imag}(z\_))^2}}$	$\frac{\lvert u0\rvert\cdot\sqrt{2}}{\sqrt{\left(\mathrm{real}\left(\sqrt{\frac{-a\cdot l}{c}}\right)\right)^2+\left(\mathrm{imag}\left(\sqrt{\frac{-a\cdot l}{c}}\right)\right)^2+2\cdot\mathrm{imag}\left(\sqrt{\frac{-a\cdot l}{c}}\right)\cdot l\cdot\omega+l^2\cdot\omega^2}}$

$$\frac{u\_eff}{\sqrt{(\mathrm{real}(z\_))^2+(\mathrm{imag}(z\_))^2}}\Big|a<0 \text{ and } l>0 \text{ and } c>0 \text{ and } u0>0 \qquad \frac{\sqrt{2\cdot c}\cdot u0}{\sqrt{-(a-c\cdot l\cdot \omega^2)\cdot l}}$$

$$\frac{\sqrt{2\cdot c}\cdot u0}{\sqrt{-(a-c\cdot l\cdot \omega^2)\cdot l}}\Big|a=\omega^2\cdot l\cdot c-4 \text{ and } l>0 \text{ and } c>0 \qquad \frac{\sqrt{2\cdot c}\cdot u0}{2\cdot\sqrt{l}}$$

$$\frac{u\_eff}{\sqrt{(\mathrm{real}(z\_))^2+(\mathrm{imag}(z\_))^2}}\Big|a>0 \text{ and } l>0 \text{ and } c>0 \text{ and } u0>0 \qquad \frac{\sqrt{2\cdot c}\cdot u0}{\left|\sqrt{a}+\sqrt{c\cdot l}\cdot\omega\right|\cdot\sqrt{l}}$$

$$\frac{\sqrt{2\cdot c}\cdot u0}{\left|\sqrt{a}+\sqrt{c\cdot l}\cdot\omega\right|\cdot\sqrt{l}}\Big|a=\omega^2\cdot l\cdot c-4 \text{ and } l>0 \text{ and } c>0 \qquad \frac{\sqrt{2\cdot c}\cdot u0}{\left|\sqrt{c\cdot l\cdot\omega^2-4}+\sqrt{c\cdot l}\cdot\omega\right|\cdot\sqrt{l}}$$

$$\frac{\sqrt{2\cdot c}\cdot u0}{\left(\sqrt{c\cdot l\cdot\omega^2-4}+\sqrt{c\cdot l}\cdot\omega\right)\cdot\sqrt{l}} \qquad \frac{\sqrt{2\cdot c}\cdot u0}{\left(\sqrt{c\cdot l\cdot\omega^2-4}+\sqrt{c\cdot l}\cdot\omega\right)\cdot\sqrt{l}}$$

$$i\_eff(u0,c,l,\omega):=\mathrm{when}\left(\omega\le\frac{2}{\sqrt{l\cdot c}},\frac{\sqrt{2\cdot c}\cdot u0}{2\cdot\sqrt{l}},\frac{\sqrt{2\cdot c}\cdot u0}{\left(\sqrt{c\cdot l\cdot\omega^2-4}+\sqrt{c\cdot l}\cdot\omega\right)\cdot\sqrt{l}}\right) \qquad \mathit{Done}$$

$$i\_eff(1,1,1,x) \qquad \begin{cases}\frac{\sqrt{2}}{2}, & x\le 2\\ \frac{\sqrt{2}}{\sqrt{x^2-4}+x}, & x>2\end{cases}$$

The graph of the piecewise defined function:



IV.23 Show that the circuit (Fig. IV.23) consisting of an infinite number of capacitors and inductors is a high-pass filter.



Fig. IV.23

Solution:

Again we use the result obtained in problem V.6, when we calculated the resultant resistance of an infinite system of resistors. Comparing Fig. IV.6a and Fig. IV.23 it can easily be seen tat the role of R_1 is taken by the impedance of the capacitor Z_C, and the role of the resistance R_2 is taken by the impedance of the inductance Z_L. Thus in accordance with formula %o3 from this problem the impedance of the system as a whole is given by

$$Z = \frac{\sqrt{{Z_L}^2 + 4Z_C Z_L} + Z_C}{2} \quad \text{with } Z_L = i\omega L,\ Z_C = \frac{1}{i\omega C}.$$

The Maxima procedure runs very similar to the previous example:

```
(%i3)   Z[L]:%i*ω*L$  Z[C]:1/(%i*ω*C)$
        Z:(sqrt(Z[C]^2+4*Z[L]*Z[C])+Z[C])/2;
```

(%o3) $$\frac{\sqrt{\frac{4L}{C} - \frac{1}{C^2\omega^2}} - \frac{\%i}{C\omega}}{2}$$

```
(%i5)   assume(L>0,C>0,ω>0)$
        Z[abs]:cabs(Z);
```

(%o5) $$sqrt\left(\left(\sin\left(\frac{\mathrm{atan2}\left(0, \frac{4L}{C} - \frac{1}{C^2\omega^2}\right)}{2}\right)\sqrt{\left|\frac{1}{C^2\omega^2} - \frac{4L}{C}\right|} - \frac{1}{C\omega}\right)^2 + \cos\left(\frac{\mathrm{atan2}\left(0, \frac{4L}{C} - \frac{1}{C^2\omega^2}\right)}{2}\right)^2 \left|\frac{1}{C^2\omega^2} - \frac{4L}{C}\right|\right)/2$$

```
(%i6)   Z[abs]:subst(atan2(0,(4*L)/C-1/(C^2*ω^2))=0,Z[abs]);
```

(%o6) $$\frac{\sqrt{\frac{1}{C^2\omega^2} + \left|\frac{1}{C^2\omega^2} - \frac{4L}{C}\right|}}{2}$$

We have again to distinguish two cases for the absolute value under the square root:

```
(%i7)   ratsimp(sqrt(1/(4*C^2*ω^2)+(1/(C^2*ω^2)-(4*L)/C)/4));
```

$$(\%o7)\quad \frac{\sqrt{1-2\,C\,L\,\omega^2}}{\sqrt{2}\,C\,\omega}$$

```
(%i8)   ratsimp(sqrt(1/(4*C^2*ω^2)-(1/(C^2*ω^2)-(4*L)/C)/4));
```

$$(\%o8)\quad \frac{\sqrt{L}}{\sqrt{C}}$$

The circuit analysed is thus a high-pass filter.
We define the impedance as a piecewise defined function and plot for $L = C = 1$.

```
(%i9)   Zabs(ω):=if ω<=1/(2*sqrt(L*C)) then
                 sqrt(1-2*C*L*ω^2)/(sqrt(2)*C*ω) else
                 sqrt(L)/sqrt(C);
```

$$(\%o9)\quad \mathrm{Zabs}(\omega):=\mathrm{if}\ \ \omega <= \frac{1}{2\sqrt{L\,C}}\ \ \mathrm{then}\ \ \frac{\sqrt{1-2\,C\,L\,\omega^2}}{\sqrt{2}\,C\,\omega}\ \ \mathrm{else}\ \ \frac{\sqrt{L}}{\sqrt{C}}$$

```
(%i10)  plot2d(subst([L=1,C=1],Zabs(ω)),[ω,0.1,2],[y,0,10]);
```



Exercise: Analyse the properties of the filtering circuits given below.



IV.24 An alternating voltage of the form $U_{in} = U_0\,(\sin(\omega\, t) + \sin(2\omega\, t)$ is applied to the circuit given in Fig. IV.24a. Calculate the ratio of the output voltage to the input voltage.



Fig. IV.24a

Solution:

The given circuit is equivalent to the circuit presented in Fig. IV.24b.



Fig. IV.24b

From the voltage divider property it can be concluded that voltage U_A and the output voltage satisfy the following equation

$$U_{out} = U_A \frac{R_C}{R + R_C}$$

where R_C denotes the impedance of the capacitor

$$R_C = \frac{1}{i\omega C}$$

It can also be seen from the potential divider property that U_A and U_{in} satisfy the following equation

$$U_A = U_{in} \frac{\dfrac{1}{\dfrac{1}{R_C} + \dfrac{1}{R + R_C}}}{R + \dfrac{1}{\dfrac{1}{R_C} + \dfrac{1}{R + R_C}}}.$$

We enter the equations from above:

```
(%i1)    eq1:Uout/UA=Rc/(R+Rc)$
(%i2)    eq2:UA/Uin=1/((1/Rc+1/(R+Rc)))/(R+1/(1/Rc+1/(R+Rc)))$
(%i3)    Rc:1/(%i*ω*C)$
```

In order to calculate the requested ratio we solve equations `eq1` and `eq2` for U_{in} and U_{out}:

```
(%i4)  solve([eq1,eq2],[Uout,Uin]);
```

(%o4) $$\left[\left[Uout=-\frac{\%i\,C\,R\,UA\,\omega+UA}{C^2\,R^2\,\omega^2-2\,\%i\,C\,R\,\omega-1},\ Uin=\frac{\%i\,C^3\,R^3\,UA\,\omega^3+4\,C^2\,R^2\,UA\,\omega^2-4\,\%i\,C\,R\,UA\,\omega-UA}{C^2\,R^2\,\omega^2-2\,\%i\,C\,R\,\omega-1}\right]\right]$$

```
(%i5)  factor(%);
```

(%o5) $$\left[\left[Uout=-\frac{UA\,(\%i\,C\,R\,\omega+1)}{C^2\,R^2\,\omega^2-2\,\%i\,C\,R\,\omega-1},\ Uin=\frac{UA\,\left(\%i\,C^3\,R^3\,\omega^3+4\,C^2\,R^2\,\omega^2-4\,\%i\,C\,R\,\omega-1\right)}{C^2\,R^2\,\omega^2-2\,\%i\,C\,R\,\omega-1}\right]\right]$$

```
(%i6)  Uout:-(UA*(%i*C*R*ω+1))/(C^2*R^2*ω^2-2*%i*C*R*ω-1)$
(%i7)  Uin:(UA*(%i*C^3*R^3*ω^3+4*C^2*R^2*ω^2-4*%i*C*R*ω-1))/
               (C^2*R^2*ω^2-2*%i*C*R*ω-1)$
```

Ratio $\eta=\frac{U_{out}}{U_{in}}$ can be entered in concise form as follows in %i8. `cabs` delivers its modulus.

```
(%i8)  η:Uout/Uin$
(%i9)  radcan(cabs(η));
```

(%o9) $$\frac{1}{\sqrt{C^4\,R^4\,\omega^4+7\,C^2\,R^2\,\omega^2+1}}$$

The phase shift – exactly the tangent of the phase shift – can be calculated by taking the ratio of the imaginary part of η and its real part.

```
(%i10)  ratsimp(imagpart(η)/realpart(η));
```

(%o10) $$\frac{3\,C\,R\,\omega}{C^2\,R^2\,\omega^2-1}$$

IV.25 Two circuits (*RL* and *RC*) are given below. An impulse of potential of the form.

$$E(t)=\begin{cases}E_0 & \text{for} \quad t_1\le t\le t_2\\ 0 & \text{for} \quad t>t_2\end{cases}$$

is applied to these circuits. Calculate the current in these circuits as a function of time.



Fig. IV.25a and IV.25b

Solution:

a) *RL* circuit (Fig. IV.25a)

During the flow of current through a coil an electromagnetic force is induced equal to $-L\dfrac{\mathrm{d}I}{\mathrm{d}t}$.

Taking into account all the drops of potential in the circuit, we can write the current flowing in the circuit in the form of a differential equation:

$$L\frac{\mathrm{d}I}{\mathrm{d}t}+R\,I=E(t).$$

We proceed with solving the equation for each of the three intervals of time in question.

We start with calculating the general solution:

For $t \le t_1$:

```
(%i1)   de1:L*'diff(i1(t),t)+R*i1(t)=0$
(%i2)   gsoln1:ode2(de1,i1(t),t);
```

(gsoln1) $\mathrm{i1}(t)=\%c\ \%e^{-\frac{R\,t}{L}}$

```
(%i3)   i1(t):=c1*%e^(-(R*t)/L);
```

(%o3) $\mathrm{i1}(t):=c1\ \%e^{\frac{-R\,t}{L}}$

For $t_1 \le t \le t_2$:

```
(%i4)   de2:L*'diff(i2(t),t)+R*i2(t)=E0$
(%i5)   gsoln2:ode2(de2,i2(t),t);
```

(gsoln2) $\mathrm{i2}(t)=\%e^{-\frac{R\,t}{L}}\left(\frac{E0\ \%e^{\frac{R\,t}{L}}}{R}+\%c\right)$

```
(%i6)   i2(t):=%e^(-(R*t)/L)*((E0*%e^((R*t)/L))/R+c2);
```

(%o6) $\mathrm{i2}(t):=\%e^{\frac{-R\,t}{L}}\left(\frac{E0\ \%e^{\frac{R\,t}{L}}}{R}+c2\right)$

For $t \ge t_2$:

```
(%i7)   i3(t):=c3*%e^(-(t*R)/L)$
```

In the next step we calculate the constants of integration:

```
(%i10)  eq1:i1(0)=0$  eq2: i1(t1)=i2(t1)$  eq3:i2(t2)=i3(t2)$
(%i11)  solve([eq1,eq2,eq3],[c1,c2,c3]);
```

(%o11) $[[c1=0,\ c2=-\frac{E0\ \%e^{\frac{R\,t1}{L}}}{R},\ c3=-\frac{E0\ \%e^{\frac{R\,t1}{L}}-E0\ \%e^{\frac{R\,t2}{L}}}{R}]]$

```
(%i12)  [c1:0,c2:-(E0*%e^((R*t1)/L))/R,
         c3:-(E0*%e^((R*t1)/L)-E0*%e^((R*t2)/L))/R]$
```

Resuming we have

```
(%i13) i1(t);
(%o13) 0
(%i14) i2(t);
```

$$\text{(\%o14)}\quad \%e^{-\frac{R\,t}{L}}\left(\frac{E0\,\%e^{\frac{R\,t}{L}}}{R}-\frac{E0\,\%e^{\frac{R\,t1}{L}}}{R}\right)$$

```
(%i15) i3(t);
```

$$\text{(\%o15)}\quad \frac{\%e^{-\frac{R\,t}{L}}\left(E0\,\%e^{\frac{R\,t2}{L}}-E0\,\%e^{\frac{R\,t1}{L}}\right)}{R}$$

We may write the current at any given moment of time in a compact form:

```
(%i16) iRL(t):=if t<t1 then 0 else
              if t<t2 then E0*%e^(-(R*t)/L)/R*(%e^((R*t)/L)-%e^((R*t1)/L))
              else E0*%e^(-(R*t)/L)/R*(%e^((R*t2)/L)-%e^((R*t1)/L))$
```

b) *RC* circuit (Fig. IV.25b)

For the circuit shown in Fig. IV.24b the charge on the capacitor at time t is given by the equation

$$R\frac{\mathrm{dQ}}{\mathrm{dt}}+\frac{Q}{C}=E(t).$$

The solution of the differential equation above may be calculated in a similar way. But from a mathematical point of view this equation is of the same type as the one obtained in part (a).

Therefore replacing I by Q and L by R and R by $\frac{1}{C}$ in `%o16` we may write:

```
(%i17) Q(t):=if t<t1 then 0 else
             if t<t2 then -%e^(-t/(C*R))*E0*C*(%e^(t1/(C*R))-%e^(t/(C*R)))
             else E0*C*%e^(-t/(C*R))*(%e^(t2/(C*R))-%e^(t1/(C*R)))$
```

And further the current as derivative of the charge $Q(t)$:

```
(%i18) diff((-%e^((-t)/(C*R)))*E0*C*(%e^(t1/(C*R))-%e^(t/(C*R))),t);
```

$$\text{(\%o18)}\quad \frac{E0\,\%e^{-\frac{t}{C\,R}}\left(\%e^{\frac{t1}{C\,R}}-\%e^{\frac{t}{C\,R}}\right)}{R}+\frac{E0}{R}$$

```
(%i19) diff(E0*C*%e^((-t)/(C*R))*(%e^(t2/(C*R))-%e^(t1/(C*R))),t);
```

$$\text{(\%o19)}\quad -\frac{E0\,\%e^{-\frac{t}{C\,R}}\left(\%e^{\frac{t2}{C\,R}}-\%e^{\frac{t1}{C\,R}}\right)}{R}$$

```
(%i20) iRC(t):=if t<t1 then 0 else
              if t < t2 then (E0*%e^(-t/(C*R))*(%e^(t1/(C*R))-
                              %e^(t/(C*R))))/R+E0/R else
              -(E0*%e^(-t/(C*R))*(%e^(t2/(C*R))-%e^(t1/(C*R))))/R$
```

Graphical representation of the currents for both circuits for a set of parameters is presennted below.

```
(%i21)  [R:1,L:1,C:1,t1:1,t2:2,E0:3]$
(%i22)  plot2d(iRL(t),[t,0,10],[y,-1,3],[title,"iRL(t)"]);
(%o22)  [C:/Users/Josef/maxout4636.gnuplot]
(%i23)  plot2d(iRC(t),[t,0,10],[y,-5,5],[title,"iRC(t)"]);
(%o23)  [C:/Users/Josef/maxout4636.gnuplot]
```





Let us try another method! We can apply the Laplace transform to solve *RL*-problem.

```
(%i34)  de3:L*'diff(i(t),t)+R*i(t)=E0*(unit_step(t-1)-unit_step(t-2));
```

(de3) $$\frac{d}{d\,t}\,i(t)+i(t)=3\,(\mathrm{unit\_step}(t-1)-\mathrm{unit\_step}(t-2))$$

```
(%i25)  laplace(de3,t,s);
```

(%o25) $$s\,\mathrm{laplace}(i(t),t,s)+\mathrm{laplace}(i(t),t,s)-i(0)=3\left(\frac{\%e^{-s}}{s}-\frac{\%e^{-2\,s}}{s}\right)$$

```
(%i26)  L_de3:s*I+I-0=3*(%e^(-s)/s-%e^(-2*s)/s);
```

(L_de3) $$I\,s+I=3\left(\frac{\%e^{-s}}{s}-\frac{\%e^{-2\,s}}{s}\right)$$

```
(%i27)  solve(L_de3,I);
```

(%o27) $$[\,I=\frac{\%e^{-2\,s}\,(3\,\%e^{s}-3)}{s^2+s}\,]$$

Maxima can perform the Laplace transform of the unit_step function (and the Dirac δ-function as well) but cannot do the reverse transformation. As we know that $F(s)\,e^{-as}$ gives a product containing unit_step($t-a$) I invent the respective function for doing the inverse Lapalce transform for this special case (`ilt(F(s),s,t)` is the maxima function for "regular" inverse transforms.

```
(%i28)  expand(%e^(-2*s)*(3*%e^s-3));
```

(%o28) $$3\,\%e^{-s}-3\,\%e^{-2\,s}$$

```
(%i29)  invl(a_,f,s,t):=trigsimp(unit_step(t-a_)*ilt(f,s,t-a_))$
(%i36)  L_sol(t):=3*unit_step(t-1)-3*unit_step(t-1)*%e^(1-t)+
                  3*unit_step(t-2)*%e^(2-t)-3*unit_step(t-2)$
(%i37)  plot2d(L_sol(t),[t,0,10],[y,-1,3],
              [title,"iRL(t) with LaplaceTr"]);
(%o37)  [C:/Users/Josef/maxout1492.gnuplot]
```

```
(%i36)  L_sol(t):=3*unit_step(t-1)-3*unit_step(t-1)*%e^(1-t)+
                  3*unit_step(t-2)*%e^(2-t)-3*unit_step(t-2)$
(%i37)  plot2d(L_sol(t),[t,0,10],[y,-1,3],
            [title,"iRL(t) with LaplaceTr"]);
(%o37)  [C:/Users/Josef/maxout1492.gnuplot]
```



We will solve this problem with TI-NspireCAS:

a) RL–circuit

deSolve$(l\cdot i1'+r\cdot i1=0,t,i1)$ ▸ $i1=c21\cdot e^{\frac{-r\cdot t}{l}}$

considering the initial condition i1(t=0)=0:

deSolve$(l\cdot i1'+r\cdot i1=0$ and $i1(0)=0,t,i1)$ ▸ $i1=0$

$i1(t):=0$ ▸ *Done*

for $t_1 \le t \le t_2$:

desolve$(l\cdot i2'+r\cdot i2=e0,t,i2)$ ▸ $i2=c23\cdot e^{\frac{-r\cdot t}{l}}+\frac{e0}{r}$

considering the condition $i2(t_1)=i1(t_1)$:

desolve$(l\cdot i2'+r\cdot i2=e0$ and $i2(t1)=i1(t1),t,i2)$ ▸ $i2=\frac{e0}{r}-\frac{e0\cdot e^{r\cdot\left(\frac{t1}{l}-\frac{t}{l}\right)}}{r}$

$i2(t):=\frac{e0}{r}-\frac{e0\cdot e^{r\cdot\left(\frac{t1}{l}-\frac{t}{l}\right)}}{r}$ ▸ *Done*

for $t > t$:

$i2(t):=\frac{e0}{r}-\frac{e0\cdot e^{\cdots}}{r}$ ▸ *Done*

for $t \geq t_2$:

desolve$(l\cdot i3'+r\cdot i3=0 \text{ and } i3(t2)=i2(t2),t,i3)$ ▸ $i3=\frac{^{-}e0\cdot\left(e^{\frac{t1\cdot r}{l}}-e^{\frac{t2\cdot r}{l}}\right)\cdot e^{\frac{-r\cdot t}{l}}}{r}$

$i3(t):=\frac{^{-}e0\cdot\left(e^{\frac{t1\cdot r}{l}}-e^{\frac{t2\cdot r}{l}}\right)\cdot e^{\frac{-r\cdot t}{l}}}{r}$ ▸ *Done*

$irl(t):=\begin{cases} i1(t), & t<t1 \\ i2(t), & t<t2 \\ i3(t), & t\geq t2 \end{cases}$ ▸ *Done*

after fixing the parameter values the above functions will change to the special solutions.

$i2(t):=\frac{e0}{r}-\frac{e0\cdot e^{\cdots}}{r}$ ▸ *Done*

for $t \geq t_2$:

desolve$(l\cdot i3'+r\cdot i3=0 \text{ and } i3(t2)=i2(t2),t,i3)$ ▸ $i3=3\cdot(e-1)\cdot e^{1-t}$

$i3(t):=\frac{^{-}e0\cdot\left(e^{\frac{t1\cdot r}{l}}-e^{\frac{t2\cdot r}{l}}\right)\cdot e^{\frac{-r\cdot t}{l}}}{r}$ ▸ *Done*

$irl(t):=\begin{cases} i1(t), & t<t1 \\ i2(t), & t<t2 \\ i3(t), & t\geq t2 \end{cases}$ ▸ *Done*

$r:=1:l:=1:t1:=1:t2:=2:e0:=3$ ▸ 3

$irl(t)$ ▸ $\begin{cases} 0, & t<1 \\ 3-3\cdot e^{1-t}, & 1\leq t<2 \\ 3\cdot(e-1)\cdot e^{1-t}, & t\geq 2 \end{cases}$

compare the result for i3!



The RC–Circuit after fixing the parameters:

$\mathbf{irc}(t):=\frac{d}{dt}(\mathbf{irl}(t))|l=r$ and $r=\frac{1}{\mathbf{c}}$ ▸ *Done*

$$\mathbf{irc}(t) \blacktriangleright \begin{cases} 0, & t<1 \\ 3\cdot \mathbf{e}^{1-t}, & 1<t<2 \\ -3\cdot(\mathbf{e}-1)\cdot \mathbf{e}^{1-t}, & t>2 \end{cases}$$

$\mathbf{c}:=1:\mathbf{t1}:=1:\mathbf{t2}:=2:\mathbf{e0}:=3$ ▸ 3



IV.26 A coil of inductance L, a resistor of resistance R and a capacitor of capacity C are parallel connected to an alternating electromotive force E (Fig. IV.26). Find the power generated across the resistor as a function of the frequency ω of the source.



Fig. IV.26

Solution:

The power generated across the resistor R is given by the expression $P=\frac{|U|^2}{R}$ where U denotes the potential across the resistor. From the law regarding a voltage divider, we obtain

$$U = E\frac{Z}{Z+r}$$

where Z denotes the resultant impedance of the elements R, L, and C respectively connected parallel:

$$\frac{1}{Z}=\frac{1}{R}+\frac{1}{R_C}+\frac{1}{R_L},\ (R_C=\frac{1}{i\omega C},\ R_L=i\omega L).$$

In the first step we find the resultant impedance.

```
(%i3)    eq:1/Z=1/R+1/R[C]+1/R[L]$
         R[C]:1/(%i*ω*C)$   R[L]:%i*ω*L$
(%i4)    ev(solve(eq,Z));
```

$$(\%o4)\quad [Z=\frac{L\,R\,\omega}{\%i\,C\,L\,R\,\omega^2+L\,\omega-\%i\,R}]$$

The solution obtained is assigned to Z. Then we evaluate the power P.

```
(%i5)    Z:(L*R*ω)/(%i*C*L*R*ω^2+L*ω-%i*R)$
(%i7)    U:E*Z/(Z+r)$
         P:factor(cabs(U)^2/R);
```

$$(P)\quad \frac{E^2\,L^2\,R\,\omega^2}{C^2\,L^2\,R^2\,r^2\,\omega^4-2\,C\,L\,R^2\,r^2\,\omega^2+L^2\,r^2\,\omega^2+2\,L^2\,R\,r\,\omega^2+L^2\,R^2\,\omega^2+R^2\,r^2}$$

We ask for the extremal values of P by applying its first derivative.

```
(%i8)    solve(diff(P,ω)=0,ω);
```

$$(\%o8)\quad [\omega=\frac{\%i}{\sqrt{C}\sqrt{L}},\ \omega=-\frac{1}{\sqrt{C}\sqrt{L}},\ \omega=-\frac{\%i}{\sqrt{C}\sqrt{L}},\ \omega=\frac{1}{\sqrt{C}\sqrt{L}},\ \omega=0]$$

From %o8 we see that only one solution makes sense: $\omega=\frac{1}{\sqrt{LC}}$.

We calculate the maximal power P which is reached for $\omega=\frac{1}{\sqrt{LC}}$.

```
(%i9)    ratsimp(subst(ω=1/(sqrt(C)*sqrt(L)),P));
```

$$(\%o9)\quad \frac{E^2\,R}{r^2+2\,R\,r+R^2}$$

For the limit values of the frequency the generated power tends to zero when frequency tends either to zero or to infinity.

```
(%i10)   [limit(P,ω,0),limit(P,ω,inf)];
(%o10)   [0,0]
(%i11)   plot2d(subst([E=1,C=1,R=1,L=1,r=1],P),[ω,0,8],[y,-0.1,0.4]);
(%o11)   [C:/Users/Josef/maxout4344.gnuplot]
```

The graphical representation of the generated power is shown below.



IV.27 A potential of the form $U_{in} = U_0 \sin^3(\omega_0 t)$ is applied to the circuit given in Fig. IV.27. The parameters of the circuit satisfy the equations $\omega_0^2 LC = 1$ and $\omega_0 LC = 2$. Calculate the transmission coefficient of the filter.



Fig. IV.27

Solution:

The exit potential in the circuit is given by $U_{out} = U_{in} \dfrac{R_L + R_C}{R + R_L + R_C}$ where

$$R_L = i\omega L,\ R_C = \frac{1}{i\omega C} \text{ and } U_{in} = U_0 \sin^3(\omega_0 t).$$

We let know all what we are knowing our CAS.

```
(%i4)    R[L]:%i*ω*L$   R[C]:1/(%i*ω*C)$
         U[in]:U[0]*sin(ω*t)^3$
         U[out]:U[in]*(R[L]+R[C])/(R+R[L]+R[C])$
```

The transmission coefficient is defined to be the ratio between the exit signal and the input signal.

```
(%i5)    ratio:U[out]/U[in]$
```

We have to calculate its modulus for the given parameters satisfying the following relations:

$$\omega_0^2 LC = 1,\ \omega_0 RC = 2 \text{ i.e. } L = \frac{1}{\omega_0^2 C},\ C = \frac{2}{\omega_0 R}.$$

```
(%i6)    ratio:subst([L=1/(C*ω0^2),C=2/(R*ω0)],ratio)$
(%i7)    ratio:factor(ratio);
```

$$\text{(ratio)}\quad \frac{\%i\,(\omega 0-\omega)\,(\omega 0+\omega)}{\%i\,\omega 0^2 - 2\,\omega\,\omega 0 - \%i\,\omega^2}$$

It can be seen from expression `(ratio)` that at frequency $\omega = \omega_0$ no potential is transferred.

One should ask which frequencies are transmitted through the circuit? To answer this question we try to calculate coefficients of Fourier series of the input potential. We notice that U_{in} is an odd function, thus it is sufficient to calculate integrals of the following form

```
(%i9)  assume(R>0,ω>0,ω0>0)$
       'integrate(sin(ω*t)^3*sin(n*ω*t),t,0,2*%pi/w)=
        integrate(sin(ω*t)^3*sin(n*ω*t),t,0,2*%pi/ω);
```

$$\text{(\%o9)}\quad \int_0^{\frac{2\pi}{w}} \sin(t\,\omega)^3 \sin(n\,t\,\omega)\,dt = -\frac{6\sin(2\,\pi\,n)}{(n^4-10\,n^2+9)\,\omega}$$

```
(%i10) factor(%);
```

$$\text{(\%o10)}\quad \int_0^{\frac{2\pi}{w}} \sin(t\,\omega)^3 \sin(n\,t\,\omega)\,dt = -\frac{6\sin(2\,\pi\,n)}{(n-3)\,(n-1)\,(n+1)\,(n+3)\,\omega}$$

It is easy to notice that the above result is equal to zero for $n > 3$ because the nominator is zero for any integer number n, whereas the denominator becomes zero for $n > 3$.

It turns out that the input signal is composed of only two frequencies: $\omega = \omega_0$ and $\omega = 3\omega_0$.

As there is no potential transmission at $\omega = \omega_0$ it is sufficient to calculate the modulus and the tangent of the phase shift of the transmission coefficient for the frequency $\omega = 3\omega_0$.

```
(%i12) ABS:cabs(ratio)$
       subst(ω=3*ω0,ABS);
```

$$\text{(\%o12)}\quad \frac{4}{5}$$

```
(%i14) ARG:imagpart(ratio)/realpart(ratio)$
       subst(ω=3*ω0,ARG);
```

$$\text{(\%o14)}\quad \frac{3}{4}$$

Exercise: Solve the above exercise for input signals of the following forms:

$$U_0 \sin^5(\omega_0\, t) \text{ and } U_0 \sin^3(2\omega_0\, t).$$

All files are available on request.

nojo.boehm@pgv.at

leon.magiera@wp.pl